


An
AMSTRAD
CPC464
Compendium

Martin Gandoff
&
Robin Kinge



An Amstrad CPC464 Compendium

An Amstrad CPC464 Compendium

Martin Gandoff and Robin Kinge



ADDISON-WESLEY PUBLISHING COMPANY

*Wokingham, England • Reading, Massachusetts • Menlo Park, California
Don Mills, Ontario • Amsterdam • Sydney • Singapore • Tokyo
Mexico City • Bogota • Santiago • San Juan*

© 1984 Newtech Publishing Ltd

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system, or transmitted in any form or by any means, electronic, mechanical, photocopying, recording or otherwise, without the prior written permission of the publisher.

The programs presented in this book have been included for their instructional value. They have been tested with care but are not guaranteed for any particular purpose. The publisher does not offer any warranties or representations, nor does it accept any liabilities with respect to the programs.

To the best of the publisher's knowledge, the contents of this book in no way transgress any other copyrighted material. Any similarity in idea or program is purely coincidental.

Cover illustration by Chris Smithers
Photographs by Tim Humphrey

Print and binding by HGA Printing Company Ltd, Brentford, Middlesex

ISBN 0-201-15439-0
ABCDE 8987654

British Library Cataloguing in Publication Data

Gandoff, Martin

An Amstrad CPC464 Compendium.

1. Computer games 2. Amstrad computer - Programming

I. Title II. Kinge, Robin

794.8'028'5404 GV1469.2

ISBN 0-201-15439-0

CONTENTS

<i>Preface</i>		vii
<i>Chapter 1</i>	How to Make Use of the Amstrad	1
<i>Chapter 2</i>	Programming Hints and "Locomotive" BASIC	3
<i>Chapter 3</i>	Computer "Art"	15
	Animation	15
	Rotation	26
	The Drawing Program	29
<i>Chapter 4</i>	Games of Skill	35
	Wordslide	37
	Reverse	42
	Flasher	45
	Fruit Machine	50
	Pennies	59
	Leapfrog	63
	Twister	68
	Mummies' Revenge	72
<i>Chapter 5</i>	"Weapon" Games	79
	Leveller	81
	Strafe	86
	Mortar	93
<i>Chapter 6</i>	Modern Games	101
	Batter	103
	Colours	109
	Demon Driver	114
	Docking	118

<i>Chapter 7</i>	Traditional Games and Sports	123
	Horse Race	125
	Os and Xs	130
	Anagram	137
	Hangman	142
<i>Chapter 8</i>	Two-user Games	151
	Moon Base	153
	Light-bikes	159
	Shoot-out	164

ABOUT THE AUTHORS

Martin Gandoff has been involved with data processing since 1966 in business and technology. He has lectured at a number of State and private colleges and is a Senior Lecturer in Systems Analysis Design at Slough College of Higher Education.

Robin Kinge is a freelance programmer with a wide knowledge of computers and computer games.

PREFACE

We have made use of the standard AMSTRAD CPC464 with its on-board cassette recorder and the CTM 640 colour monitor for the development of this book and, in all respects, we think the Amstrad is a very nice machine for writing games. It is great fun to use and has lots of facilities.

This book shows a good range of games for you to key in and play and we hope you will want to make changes to them so as to allow for your own variations. Before long, you should be able to start writing your own games programs.

The first chapter of the book will help you enter and run a particular game in the most efficient way and to sort out some of the mistakes that you are bound to make. Some of the games programs are quite long, and even the most careful programmer can make a mistake.

The next chapter tells you something about the “Locomotive” BASIC used in the games and explains the way we have written the programs using its very powerful statements and commands.

Chapter 3 introduces the machine for graphics, animation and art. First there are some animated “art” displays which show you just how nice a display you can get from a very small program. Then for those who want to add scenes to games, there is a clever little program that allows you to generate “solid” shapes.

Next, we introduce you to a drawing program that can be used to paint very detailed pictures. Once you are familiar with writing games, you might be able to use this to build up really nice backgrounds, perhaps for adding to adventure games.

The rest of the book contains lots of games grouped into different types. We hope that you will try all of them, and we expect you to get as much enjoyment out of playing the games as we did writing them.

Finally, our thanks to Mark Hopgood for writing many of the games and sorting out the programs into their final form.

Martin Gandoff

Robin Kinge

November 1984

HOW TO MAKE USE OF THE AMSTRAD

Introduction

Before you bought your Amstrad CPC464, you would have looked at the alternatives in the same price range. At the time of writing there are more than 10 machines of a similar price; comparing the Amstrad's wide range of facilities with the others, we think that it is one of the top three. Amstrad have adopted "Locomotive" BASIC which is fast and has powerful facilities.

The *User Guide* is a mine of information about the extensive features of the Amstrad CPC464. Our second chapter is not intended to replace the manual in any way, but we feel that you will find it useful to have a classification and explanation of statements particularly used in our games programs.

The equipment used for writing the games was:

- An Amstrad CPC464 microcomputer with the standard cassette recorder and video screen.
- An Epson FX80 printer connected through the Centronics output port.

The various programs were saved and sorted on cassette using the cassette recorder (at SPEED WRITE 1).

Entering Programs

The Amstrad keyboard has a smooth action and a very positive feel. We also found that the colour coding of the keys was useful to help in gaining familiarity.

The layout of the keyboard is good except that the cursor keys could have been better placed. It is quite easy to hit one of the keypad keys with your hand when using them.

There is no reset key as such, but the combination CONTROL, SHIFT and ESCAPE does the same job. It clears the entire memory, so be careful when using it.

The Cassette Recorder

This is the main backing store for the Amstrad CPC464. The drive SAVES and LOADS at two different rates:

SPEED WRITE 0 This SAVES and LOADS at 1000 Baud (about 120 characters per second).

SPEED WRITE 1 This SAVES and LOADs at 2000 Baud (about 250 characters per second).

SPEED WRITE 0 (which is the default option) should be used when you are transferring programs between your own Amstrad and others. This is because the slower rate ensures that it reads each character correctly.

Most commercial programs on tape will be saved at **SPEED WRITE 0**, again to ensure that the program goes in correctly.

SPEED WRITE 1 is faster than **SPEED WRITE 0** and of course loads a program approximately twice as fast. This means, however, that it is slightly less reliable. However, a program saved at **SPEED WRITE 1** will almost always reload into the machine it was saved on.

Often when you are loading programs, the tape will stop for a few seconds. Don't worry since this is quite normal behaviour. It is the Amstrad doing its tape control. You must let the tape continue running or the program will not load. The Amstrad will tell you when it has finished the load by printing **READY** or by running the program.

You load a program by pressing the **CONTROL** key next to the space bar and, while holding it down, pressing the small **ENTER** key on the numeric keypad.

The Amstrad will tell you to press **PLAY** on the cassette player and any key on the keyboard. It will then load the first program it comes to on the cassette.

The Function Keys

Unlike some computers, the Amstrad allows for the use of the numeric keypad as special function keys. They can be programmed to represent any command so as to save you the effort of having to enter it in full.

To associate a key with a command, you enter:

KEY x,cmdnd

where **x** is the number of the key to be programmed, and **cmdnd** is the command.

For example:

KEY 1, "RUN"

Whenever key 1 on the keypad is pressed, **RUN** will appear on the screen and the command will be executed as soon as you press **ENTER**.

A useful extension is to make the command "auto-active". You establish the command with:

KEY 1, cmdnd + CHR\$(13)

The **CHR\$(13)** is the internal code equivalent for **ENTER**.

PROGRAMMING HINTS AND “LOCOMOTIVE” BASIC

Program Structure

The first thing you will probably notice when you look at the programs we have written is that they seem to be very long indeed. You will also notice that the reason for this is that most lines are coded as single statements.

There are two major reasons for this. First of all, people have got into the habit of making their BASIC take up as little room as possible, because it is more efficient for the translator program as many games computers have a limited amount of memory. The Amstrad has a good-sized memory and you need not worry about there not being enough space available to translate your BASIC into machine code. (The games programs generally use well under a quarter of the available memory space.)

The other reason is that people have become lazy and seem to prefer saving a little time when entering programs, at the expense of making them readable. We have spaced the program lines out and we hope that you will take note of what we have done and will enter the program lines in the format in which they are shown in the book. Of course, if you are desperately short of time, you could leave out some of the spaces. For example, you will see many cases where we have “indented” lines, inside a FOR...NEXT for example. We think that this makes the program easier to read but if you cannot be bothered, then do not put them in.

If you have been clever in the past, you will already know about using GOSUB and RETURN for subroutines to make the logic of your programs easier to understand and therefore easier to program.

A unique aspect of Amstrad BASIC is the facility to handle timer interrupts. The machine has four internal clocks and each can be nominated to allow for an activity to be carried out **every** so often. The statement takes the form:

```
EVERY n,c GOSUB line-number
```

Each and every n time units ($\frac{1}{50}$ th of a second) of clock c , the system causes the subroutine at “line-number” to be executed. You can get very clever by making use of all four clocks if you wish.

The “Graphics” Statements

Now let us look at what we will call the graphics statements. We can divide these into two main groups, as follows:

- Statements relating to origins, windows and colour.
- The drawing statements themselves.

MODE

We must mention this first because many others depend on it. The statement sets the resolution of the screen and also the number of colours that can be used.

MODE n			
<i>n value</i>	<i>text area</i>	<i>pixels</i>	<i>colours</i>
0	20 × 25	160 × 200	16
1	40 × 25	320 × 200	4
2	80 × 25	640 × 200	2

ORIGIN

This enables you to reset the **graphics origin**, i.e. the origin from which the drawing statements are located. It takes the form:

ORIGIN x,y{l,r,t,b}

The x,y specifies the actual location of the origin in pixels from the default origin. The term in the curly brackets enables you to specify a graphics window. The l specifies the number of pixels from the left of the screen that the left-hand side of the window will be, r the number from the left for the right-hand side and t and b specify the number of pixels that the top and bottom will be from the top and the bottom of the screen, respectively.

We will have more to say about windows shortly.

LOCATE

This associated statement enables text to be located other than at the current text cursor position.

LOCATE x,y

We are asking the cursor to be relocated at character x on line y, where character size and line spacing are controlled by MODE.

MOVE

MOVE x,y

This statement is discussed under drawing statements on page 9, although it is mentioned here since it is also connected with relocating origins.

TEST

This enables us to look at a particular pixel (x,y) and determine which PEN colour is associated with it:

```
TEST x,y
```

The x,y specifies the required pixel.

TAG/TAGOFF

If we want to label a diagram, we can arrange to make text appear at the current graphics cursor position. After the most recent graphics statement, we say:

```
TAG
```

The next PRINT statement will cause printing as we have asked and later prints will follow on until we give a:

```
TAGOFF
```

which returns any further text to the current cursor position.

Windows

Now let us look at the concept of windows. These are selected areas on the screen with which text can be associated.

The BASIC statement to define a window is:

```
WINDOW # n,l,r,t,b    n = 0-7
```

n specifies the stream the window is associated with so that it can be referred to in PRINT, LOCATE and other statements. A stream can be numbered 0-7.

Having specified a stream, we can say for example:

```
100 PRINT #2, "IN WINDOW 2"
110 PRINT #3, "IN WINDOW 3"
```

The l, r, t and b specify the sides of the window (as we explained for a window defined using the option in ORIGIN) except that they are specified in characters determined by the current MODE. For example:

```
100 MODE 1
200 WINDOW #3, 1, 10, 1, 10
```

sets up a window with its top left-hand corner in the top left-hand corner of the screen and its bottom right-hand corner 10 characters down and 10 characters across.

WINDOW SWAP

When you do not specify a stream to be associated with PRINT, LOCATE etc., the default value is assumed to be 0. It is possible to alter this default as follows:

WINDOW SWAP s1, s2

This changes the significance of a stream from s1 to s2. In other words, if you have said:

WINDOW SWAP 0,4

the stream assumed, if none is specified, becomes 4.

Similarly, if you have said:

WINDOW SWAP 2,5

on execution of:

PRINT #2, "FRED"

the output text will actually appear in the window associated with stream 5.

Scrolling

There is no separate statement to produce scrolling, but you can get clever with CHR\$.

You position the text cursor to the top left, with LOCATE, and then say:

PRINT #n,CHR\$(10)

This will scroll the screen down by one line.

If you LOCATE the bottom left, you can:

PRINT #n,CHR\$(11)

to scroll up by one line.

CLS

This enables you to clear out a window to the current PAPER colour (colours will be discussed shortly).

CLS #n

where n is the specified stream. If you do not specify a stream, the screen associated with stream 0 is cleared.

Colours

The approach to colours with the Amstrad is flexible and is very good for games. Essentially there are 27 colours to choose from, numbered 0-26. In practice, the screen can only handle a maximum of 16 colours which must be selected from these, i.e. only 16 colours can be in use at the same time.

INK

This enables the 16 **logical** colours to be selected from the 27 **actual** colours.

Details of the actual colours and the values selected from them for the default logical colours can be obtained from the *Amstrad User Guide*.

INK l,a

where l specifies the logical colour (0-15) and a the actual colour (0-27).

PEN

Having selected the logical colours (which need not all be different, by the way), you select the logical colour in which text will appear:

PEN l

All later PRINT statements will appear in the selected window in the logical colour denoted by l. A further PEN will change the logical colour of the text.

PAPER

The same as PEN, but the colour of the window background is controlled:

```
100 INK 3,20    Bright Cyan
110 INK 6,11    Sky Blue
120 INK 9,5     Mauve
140 PEN 3
150 PAPER 6
160 PRINT "HELLO"
170 PAPER 9
180 PRINT "GOODBYE"
```

HELLO will appear in bright cyan on sky blue paper, while GOODBYE will be bright cyan on mauve.

If you then give a new INK statement, i.e. changing the significance of a logical colour, then anything on the screen produced by specifying PEN or PAPER in the old logical colour will be changed to the new logical colour.

BORDER

This enables you to set up a coloured border around the screen:

BORDER c

where c is the logical colour.

Flashing

By specifying another colour with INK, it is possible to arrange for a flashing effect. Also, the rate of flashing can be controlled with SPEED INK.

INK l,a1,a2

where l is the logical colour formed by alternate flashing of actual colours a1 and a2.

10 INK 3,20,6
20 INK 12,9,1
30 PEN 3
40 PAPER 12
50 PRINT "HELLO"

This gives text flashing between bright cyan and red on flashing green on blue paper.

SPEED INK

This is used to control the flashing rate:

SPEED INK f1,f2

where f1 and f2 specify the units of flash for each component in units of 1/50 second, so:

SPEED INK 50,100

would cause a sequence of a one-second flash followed by a two-second flash in the other colour.

Flashing can also be induced in a border:

BORDER 20,9
SPEED INK 25,25

.....

The default value is 20,20, i.e. flash components of 2/5ths of a second.

User-defined Graphic Characters (UDGs)

These are characters you build up yourself. You make up the character using an 8x8 grid and then use the following statement:

SYMBOL n, d1,d2,. . . ,d8

where n is the decimal equivalent of the ASCII code denoting the "official" character you are defining as a UDG, and the d1-d8 are the decimal equivalents of the bit strings you have made up from your grid.

Try this one out:

20 SYMBOL 91,24,60,219,255,241,36,102,129
30 PRINT CHR\$(91)

In order to stop you mistakenly re-using the symbol number for another UDG, you can protect it by saying:

SYMBOL AFTER n

where n is the symbol number. Any attempt to re-use any symbol less than or equal to n will be prevented.

Now let us look at the drawing statements, i.e. those that are specifically used to produce geometric shapes.

The Drawing Statements

PLOT

This enables you to produce a point:

```
PLOT x,y,c
```

where x,y specifies the location of the point in pixels, relative to the window origin, and c is an optional new colour for succeeding graphics statements (something like PEN for graphics).

DRAW

This allows a line to be drawn from the current graphics cursor location (or the last plotted point) to the specified point:

```
DRAW x,y,c
```

As with PLOT, c specifies a new colour.

For PLOT and DRAW, if a new colour is specified, it has the effect that a PEN statement has on text, i.e. it changes the graphics colour until a further DRAW or PLOT is used with a new colour.

MOVE

This enables you to move the graphics cursor from the current position or the last plotted point or line:

```
MOVE x,y
```

No change of colour is possible with MOVE.

A word of advice

When you have finished running a game and want to start entering or loading another one, **make sure that you reset, with the combination CONTROL/SHIFT/ESCAPE, before continuing.** If you do not, then a number of options may have been set by the earlier game that you will not want. The commands that you may need to enter to “clean things up” before you carry on could include any of the following:

```
INK
BORDER
PEN
NEW
PAPER
WINDOW
```

All of these will be “cleared” to their default values when you reset.

Sound

We can't finish this chapter without saying something about the sound facilities on the Amstrad. They are really superb and can be accomplished with the SOUND statement. This is very complicated and we will just deal with the basic four parameters:

SOUND channel pitch duration volume

- The channel is 1, 2 or 4.
- Pitch starts at 0 and goes down to 4095.
- Duration is in steps of milliseconds from 0 to 32767.
- Volume is on a loudness scale of 0-15.

Information on notes and pitches can be obtained from the *Amstrad User Guide*.


To help you understand SOUND and to have some fun, we have written a demonstration program called SYNTH 464.

This program allows you to alter the pitch, duration and volume of a given note. It also allows you to mix in one ENVELOPE statement. Envelopes are covered in great detail in the *User Guide*.

The parameters to be altered are selected using the cursor controls, while the up and down controls in combination with the shift key alter the actual values.

```

100 REM *****
110 REM * Synth 464 *
120 REM *****
130 p$="IQIWIEIRITIIYUIIIOIPI"
    
```



```
140 p1$="12131*1516171*19101"
150 INK 0,0:INK 1,25:INK 2,2:INK 3,17
160 MODE 1
170 PEN 1
180 LOCATE 16,1:PRINT "Synth 464"
190 PEN 2
200 PRINT
210 PRINT "this is a very simple sort of
"
220 PRINT "monophonic synthesiser."
230 PRINT
240 PRINT "the following keys are used f
or the "
250 PRINT "keyboard..."
260 PRINT
270 PRINT " ";p1$
280 PRINT " ";p$
290 PRINT
300 PRINT "the envelope settings at the
top of the"
310 PRINT "screen correspond with the fi
rst 15 "
320 PRINT "parameters in the manual."
330 PRINT "These parameters can be chang
ed by"
340 PRINT "hitting the space bar, and th
en using"
350 PRINT "the cursor control keys to mo
ve the"
360 PRINT "cursor around."
370 PRINT "To change a parameter use <SH
IFT> and "
380 PRINT "either the up or down cursor
keys."
390 PEN 3
400 PRINT
410 PRINT " Press any key for Synth 46
4"
```

```
420 IF INKEY$="" THEN 420
430 CLS
440 INK 0,11
450 INK 2,26
460 INK 1,0
470 PEN 1
480 c1$=SPACE$(40)
490 DIM a(2,9),lm(2,9)
500 FOR i=1 TO 9
510 READ l,b
520 FOR j=1 TO 2
530 lm(j,i)=l
540 bm(j,i)=b
550 a(j,i)=1
560 LOCATE i*4,j*2:PRINT a(j,i)
570 NEXT j
580 NEXT i
590 LOCATE 20,18:PRINT "duration:-"
600 LOCATE 1,2:PRINT "ENV"
610 LOCATE 1,4:PRINT "ENT"
620 REM *****
630 REM * Wait for a key *
640 REM *****
650 a$=INKEY$
660 IF a$="" THEN GOTO 650
670 PEN 1
680 LOCATE 9,15:PRINT p$
690 LOCATE 10,14:PRINT p1$
700 PEN 2
710 IF a$="q" THEN p=478:LOCATE 10,15:PR
INT CHR$(143)
720 IF a$="2" THEN p=451:LOCATE 11,14:PR
INT CHR$(143)
730 IF a$="w" THEN p=426:LOCATE 12,15:PR
INT CHR$(143)
740 IF a$="3" THEN p=402:LOCATE 13,14:PR
INT CHR$(143)
750 IF a$="e" THEN p=379:LOCATE 14,15:PR
INT CHR$(143)
```

```

760 IF a$="r" THEN p=358:LOCATE 16,15:PR
INT CHR$(143)
770 IF a$="s" THEN p=338:LOCATE 17,14:PR
INT CHR$(143)
780 IF a$="t" THEN p=319:LOCATE 18,15:PR
INT CHR$(143)
790 IF a$="6" THEN p=301:LOCATE 19,14:PR
INT CHR$(143)
800 IF a$="y" THEN p=284:LOCATE 20,15:PR
INT CHR$(143)
810 IF a$="7" THEN p=268:LOCATE 21,14:PR
INT CHR$(143)
820 IF a$="u" THEN p=253:LOCATE 22,15:PR
INT CHR$(143)
830 IF a$="i" THEN p=239:LOCATE 24,15:PR
INT CHR$(143)
840 IF a$="9" THEN p=225:LOCATE 25,14:PR
INT CHR$(143)
850 IF a$="o" THEN p=213:LOCATE 26,15:PR
INT CHR$(143)
860 IF a$="0" THEN p=201:LOCATE 27,14:PR
INT CHR$(143)
870 IF a$="p" THEN p=190:LOCATE 28,15:PR
INT CHR$(143)
880 PEN 3
890 IF a$=CHR$(240) AND dur>0 THEN dur=d
ur-5:LOCATE 30,18:PRINT dur;" ":GOTO 650
900 IF a$=CHR$(241) AND dur<150 THEN dur
=dur+5:LOCATE 30,18:PRINT dur;" ":GOTO 6
50
910 PEN 1
920 IF a$=" " THEN GOTO 980
930 SOUND 1,p,dur,15,1,1
940 GOTO 650
950 REM *****
960 REM * Envelope adustment *
970 REM *****
980 x=1:y=1
990 a$=INKEY$

```



```

1000 IF a$="" THEN 990
1010 LOCATE y*4+1,x*2+1
1020 PRINT " "
1030 IF a$=CHR$(240) AND x>1 THEN x=1
1040 IF a$=CHR$(241) AND x<2 THEN x=2
1050 IF a$=CHR$(242) AND y>1 THEN y=y-1
1060 IF a$=CHR$(243) AND y<9 THEN y=y+1
1070 IF a$=CHR$(245) AND a(x,y)>bm(x,y)
THEN a(x,y)=a(x,y)-1
1080 IF a$=CHR$(244) AND a(x,y)<lm(x,y)
THEN a(x,y)=a(x,y)+1
1090 IF a$=" " THEN 1140
1100 LOCATE y*4,x*2:PRINT a(x,y);" "
1110 LOCATE y*4+1,x*2+1
1120 PRINT CHR$(143)
1130 GOTO 990
1140 FOR i=1 TO 9
1150 FOR j=1 TO 2
1160 LOCATE i*4,j*2:PRINT a(j,i);" "
1170 NEXT j
1180 NEXT i
1190 ENV 1,a(1,1),a(1,2),a(1,3),a(1,4),a
(1,5),a(1,6),a(1,7),a(1,8),a(1,9)
1200 ENT 1,a(2,1),a(2,2),a(2,3),a(2,4),a
(2,5),a(2,6),a(2,7),a(2,8),a(2,9)
1210 GOTO 650
1220 a$=INKEY$:IF a$="" THEN 1220 ELSE P
RINT ASC(a$):GOTO 1220
1230 DATA 217,0,127,-128,255,0
1240 DATA 217,0,127,-128,255,0
1250 DATA 217,0,127,-128,255,0

```

Now we can start looking at the games programs.

Chapter 3


COMPUTER “ART”

In this chapter, we will first show how nice you can make displays appear with some quite simple programs.

Animation

Just enter and run COIL, SPIRAL, SPINNING PLANETS, CYLINDERS, 3D-CUBES, SEASICK and LINES and then sit back and watch. Then look at the programs and see how it is done. You will learn a lot about advanced Locomotive statements this way.

```
100 REM *****
110 REM * coil *
120 REM *****
130 RAD
140 FOR i=1 TO 15:INK i,i:NEXT i
150 INK 2,0
160 MODE 0
170 ic=0
180 id=1
190 CLS
200 FOR n=602 TO 0 STEP -6
210 z1=SIN(n)*1.2
220 z2=COS(n)*1.3
230 x=255+(n/3)*z1
240 y=200+(n/3)*z2
250 r=n/12
260 GOSUB 420
270 id=id+1:IF id>=16 THEN id=id-15
```



```

280 ic=id
290 NEXT n
300 a=1
310 b=2
320 a=a+1:IF a=16 THEN a=1
330 b=b+1:IF b=16 THEN b=1
340 INK a,0
350 INK b,6
360 FOR i=1 TO 50
370 NEXT i
380 GOTO 320
390 REM *****
400 REM * Draw rough circle *
410 REM *****
420 ORIGIN x,y
430 FOR i=0 TO 180 STEP 20
440 ax=r*COS(i)
450 ay=r*SIN(i)
460 MOVE -ax,ay
470 DRAW ax,ay,ic
480 MOVE -ax,-ay


```

```

490 DRAW ax,-ay,ic
500 NEXT i
510 RETURN


100 REM *****
110 REM * Spiral *
120 REM *****
130 DEG
140 MODE 0
150 ic=0
160 id=1
170 CLS
180 FOR n=602 TO 0 STEP -7
190 z1=SIN(n)*1.2
200 z2=COS(n)*1.3
210 x=255+(n/3)*z1
220 y=200+(n/3)*z2
230 r=n/12
240 GOSUB 430
250 id=id+1:IF id=16 THEN id=1

```



```
260 ic=id
270 NEXT n
280 a=1
290 b=1
300 a=a+1:IF a=16 THEN a=1
310 b=a
320 FOR i=1 TO 15
330 b=b+1:IF b=16 THEN b=1
340 INK i,b
350 NEXT i
360 GOTO 300
370 INK 1,6:INK 2,3:INK 3,4
380 FOR i=1 TO 100:NEXT i
390 GOTO 280
400 REM *****
410 REM * Draw rough circle *
420 REM *****
430 ORIGIN x,y
440 FOR i=0 TO 180 STEP 20
450 ax=r*COS(i)
460 ay=r*SIN(i)
470 MOVE -ax,ay
480 DRAW ax,ay,ic
490 MOVE -ax,-ay
500 DRAW ax,-ay,ic
510 NEXT i
520 RETURN

100 REM *****
110 REM * Spinning planets *
120 REM *****
130 MODE 0
140 INK 0,0:BORDER 0
150 INK 8,2
160 GOSUB 730:REM *** draw stars
170 GOSUB 570
180 x=590:y=200:r=50
190 GOSUB 370
200 x=30:y=50:r=66:GOSUB 370
```



```

210 x=145:y=300:r=22
220 GOSUB 370
230 REM *****
240 REM * Animate *
250 REM *****
260 a=1:b=2
270 c=9:d=10
280 a=a+1:IF a=8 THEN a=1
290 b=b+1:IF b=8 THEN b=1
300 c=c+1:IF c=16 THEN c=9
310 d=d+1:IF d=16 THEN d=9
320 INK a,24:INK b,2
330 INK c,0:INK d,21
340 FOR j=1 TO 100:NEXT j
350 GOTO 280
360 REM *****
370 REM * Draw a planet *
380 REM * x=x position *
390 REM * y=y position *
400 REM * r=radius *
410 REM *****
420 ORIGIN x,y:col=1:DEG

```


```

430 FOR j=-r TO r STEP 5
440 col=col+1:IF col=8 THEN col=1
450 FOR i=tilt TO tilt+180 STEP 2
460 a=SIN(i)*j
470 b=COS(i)*r
480 PLOT a,b,col
490 NEXT i
500 NEXT j
510 RAD:FOR i=0 TO PI*2
520 a=COS(i)*r:b=SIN(i)*r
530 MOVE -a,b:DRAW a,b,8
540 NEXT i
550 RETURN
560 REM *****
570 REM * Draw line *
580 REM *****
590 x1=400:y1=200
600 col=15:k=0
610 RAD
620 FOR j=0 TO 200 STEP 7
630 k=k+0.1
640 ORIGIN x1,y1
650 a=COS(k)*(j*2):b=SIN(k)*(j*2)
660 PLOT a,b,col
670 a=(COS(k)*j)+k:b=(SIN(k)*j)+k
680 DRAW a,b,col
690 col=col-1:IF col=8 THEN col=15
700 NEXT j
710 RETURN
720 REM *****
730 REM * Draw stars *
740 REM *****
750 col =9
760 FOR y=0 TO 400 STEP 17
770 x1=INT(12*RND(TIME)+10)
780 FOR x=0 TO 640 STEP x1
790 PLOT x,y,col
800 col=col+1:IF col=16 THEN col=9

```


```
810 NEXT x
820 NEXT y
830 RETURN

100 REM *****
110 REM * Cylinders *
120 REM *****
130 MODE 0
140 DEG
150 FOR i=0 TO 15
160 INK 1,0
170 NEXT i
180 INK 3,5
190 FOR i=0 TO 360 STEP 3.21
200 a=SIN (i)*35
210 b=COS (i)*100
220 PLOT 100+a,150+b,c
230 DRAW 300+a,250+b
240 DRAW 500+a,150+b
250 c=c+1
260 IF c=16 THEN c=0
```




```
270 NEXT i
280 a=1
290 b=2
300 a=a+1
310 IF a=16 THEN a=1
320 b=b+1
330 IF b=16 THEN b=1
340 INK a,6
350 INK b,24
360 FOR j=1 TO 20
370 NEXT j
380 GOTO 300
```

```
100 REM *****
110 REM * 3D-Cubes *
120 REM *****
130 DI
140 MODE 0
150 FOR i=1 TO 15
160 INK i,i*2.1
```



```


170 NEXT i
180 INK 0,0
190 x=INT(600*RND(TIME)+1)
200 y=INT(400*RND(TIME)+1)
210 c=INT(12*RND(TIME)+1)
220 z=INT(100*RND(TIME)+1)
230 GOSUB 260
240 GOTO 190
250 REM *****
260 REM * draw a cube x,y,z *
270 REM *****
280 FOR i=1 TO z
290 PLOT x,y+i:DRAW x+z,y+i,c
300 PLOT x+i,y+z:DRAW x+i+z/3,y+z/3+z,c+
1
310 PLOT x+z,y+i
320 DRAW x+z/3+z,y+z/3+i,c+2
330 NEXT i
340 RETURN

```

```

10 REM *****
20 REM * Sea sick *
30 REM *****
40 PRINT CHR$(23);"1"
50 MODE 0
60 FOR i=0 TO 639 STEP 4
70   IF INKEY$=" " THEN 160
80   PLOT i,0
90   DRAW i,399,c
100  x=i/1.6
110  PLOT 0,x
120  DRAW 639,x
130  c=c+1
140  IF c=16 THEN c=1
150 NEXT i
160 a=1
170 b=2
180 c=1

```



```


190 a=a+1
200 IF a=16 THEN a=1
210 b=a
220 FOR i=1 TO 15
230   b=b+1
240   IF b=16 THEN b=1
250   INK b,c
260 IF i-1 =0 THEN INK 15,0:GOTO 280
270 INK i-1,0
280   c=c+1
290   IF c=16 THEN c=1
300 NEXT i
310 IF INKEY$="" THEN GOTO 190
320 GOTO 60
330 REM * Change C in line 250 to 17
340 REM * for a different effect
350 REM * Use space bar to start and
360 REM * Stop animation

```

```

100 REM *****
110 REM * Lines *
120 REM *****
130 REM
140 MODE 2
150 BORDER 0
160 DEF FNa(x)=INT(x*RND(1))+1
170 a=FNa(50)+50
180 b=FNa(100)+50
190 c=FNa(100)+50
200 d=FNa(50)+100
210 a1=FNa(20)
220 b1=FNa(20)
230 c1=FNa(20)
240 d1=FNa(20)
250 INK 0,0:PAPER 0:CLS:INK 1,26:PEN 1
260 PLOT a,b : DRAW c,d
270 a=a+a1:b=b+b1:c=c+c1:d=d+d1
280 IF a<20 OR a>620 THEN a1=-a1
290 IF b<20 OR b>380 THEN b1=-b1
300 IF c<20 OR c>630 THEN c1=-c1

```



```

310 IF d<20 OR d>380 THEN d1=-d1
320 IF INKEY$="" THEN GOTO 260
330 RUN

```

Rotation

This is a program that can produce some startling effects. The idea is to generate a figure using lines or curves made up from points and then to tell the program to rotate the figure a certain number of times to make up a solid figure.

First, you enter the number of points that will be needed to create your basic shape. Then you enter pairs of x,y coordinates in order to map out the shape so as to control the level of detail in the final display.

The program will then ask you how many rotation steps you want. It is a good idea to make this somewhere between 6 and 15 since the screen gets rather crowded with many more.


We can show you a couple of examples of what is possible. The first screen shot shows a "wine glass" and the second shows you how to create a jewel with many facets.

You can have lots of fun with this program.

```

100 REM *****
110 REM * Rotation *
120 REM *****
130 DEG


```



```

140 MODE 1
150 LOCATE 16,1:PRINT "Rotation"
160 PRINT
170 PRINT "Input the number of points yo
u"
180 PRINT
190 PRINT "want to rotate, Then the 'x'
and"
200 PRINT
210 PRINT "'y' co-ordinates of the point
s"
220 PRINT
230 PRINT "and then the number of rotati
on"
240 PRINT
250 PRINT "steps, and I will rotate"
260 PRINT
270 PRINT "your object in 3D."
280 PRINT
290 PRINT "      Press any key to start"
300 IF INKEY$="" THEN GOTO 300
310 MODE 2
320 INK 1,0:INK 0,17
330 BORDER 17
340 INPUT "How many points";p
350 DIM a(p),b(p):PLOT 0,0
360 CLS
370 REM *****
380 REM * Input points *
390 REM *****
400 FOR n=1 TO p
410 INK 4,24
420 LOCATE 1,1
430 PRINT SPACE$(159);
440 LOCATE 1,1
450 INPUT"Input x value ";a(n)
460 INPUT"Input y value ";b(n)
470 IF n<>1 THEN DRAW a(n),b(n),3

```



```

480 IF n=1 THEN PLOT a(n),b(n),3
490 NEXT n
500 LOCATE 1,1
510 PRINT SPACE$(159);
520 REM *****
530 REM * Rotate the shape *
540 REM *****
550 LOCATE 1,1
560 INPUT "Rotation steps";s
570 CLS
580 FOR n=1 TO p-1
590 FOR q=360 /s TO 360 STEP 360/s
600 z1=SIN ((q)):z2=COS((q))
610 PLOT 300+ a(n)*z1,120+b(n)+(a(n)/10
) *z2
620 DRAW 300+a(n+1)*z1,120+b(n+1)+(a(n+1
)/10)*z2
630 NEXT q
640 NEXT n
650 REM *****
660 REM * Rings *
670 REM *****

```

```

680 FOR n=1 TO p
690 PLOT 300+a(n)*z1+z1,120+b(n)+(a(n)/1
0)*z2+z2
700 FOR q=360/s TO 360 STEP 360/s
710 z1=SIN ((q)):z2=cos((q))
720 DRAW 300+a(n)*z1+z1,120+b(n)+(a(n)/1
0)*z2+z2
730 NEXT q
740 NEXT n

```

The Drawing Program

We could say a lot about what to use this program for, but all you need to know is how to control it. You will soon think up loads of ideas.

The cursor keys move the graphics cursor, and other controls are as follows:

- O Toggles between over and draw. When over is selected, the cursor does not disturb what it is moving over. When draw is used the cursor leaves a point in the current pen colour.
- R Remembers where the cursor is (fix a point).
- L Draws a line from a point remembered to the cursor.
- U Updates the display at the top of the screen.
- B Draws a box (filled or unfilled). The corners of the box are the current cursor position and a point remembered.
- C Draws a circle (filled or unfilled).
- P Changes the colour of a pen.

Now you are ready to go.

```

100 REM *****
110 REM * Draw *
120 REM *****
130 MODE 1
140 PAPER 0:INK 0,0
150 PEN 2
160 LOCATE 18,1:PRINT "Draw"
170 PEN 1
180 PRINT
190 PRINT " Cursor pad moves cursor"
200 PRINT
210 PRINT "O Toggles between over and dr
aw"
220 PRINT " When over is used the curso
r"

```



```
230 PRINT " does not disturb what it is
"
240 PRINT " moving over."
250 PRINT " When draw is used the curso
r"
260 PRINT " leaves a point in the curre
nt"
270 PRINT " pen colour."
280 PRINT "R Remembers where the cursor
is."
290 PRINT "L Draw a line from cursor to
"
300 PRINT " point remebered."
310 PRINT "U Update display at top of sc
reen."
320 PRINT "B Draw a box (filled of unfil
led)"
330 PRINT " corners of box are current
cursor"
340 PRINT " position and point remember
ed."
350 PRINT "C Draw a circle (filled or un
filled)"
360 PRINT "P Change colour of a pen."
370 PRINT
380 PRINT "The display at the top shows
x and y"
390 PRINT "co-ordinates and the current
pen colour"
400 PRINT
410 PEN 3
420 PRINT " Press any key to start"
430 PEN 1
440 IF INKEY$="" THEN GOTO 440
450 MODE 0
460 DEG
470 x=319:y=200
480 pn=2
490 c1$=SPACE$(39)
```

```

500 REM *****
510 REM * Update screen *
520 REM *****
530 LOCATE 1,1:PRINT c1$
540 LOCATE 1,1:PRINT "x=";x;" " :LOCATE 1
0,1:PRINT "y=";y;" "
550 LOCATE 16,1:PRINT pn
560 PLOT x,y,RND(TIME)*15
570 REM *****
580 REM * Check keyboard *
590 REM *****
600 a$=INKEY$
610 IF o=1 THEN PLOT x,y,15-cc:PLOT x,y,
cc
620 IF o=0 THEN PLOT x,y,pn
630 IF a$="" THEN GOTO 600
640 y=y+(a$=CHR$(241) AND y>2)*2-(a$=CHR
$(240) AND y<383)*2
650 x=x+(a$=CHR$(242) AND x>2)*4-(a$=CHR
$(243) AND x<649)*4
660 IF a$="o" THEN o=1-o:GOTO 740
670 pn=pn+(a$="a" AND pn>0)-(a$="d" AND
pn<15)
680 IF a$="r" THEN x1=x:y1=y
690 IF a$="l" THEN DRAW x1,y1,pn
700 IF a$="u" THEN 530
710 IF a$="b" THEN GOTO 810
720 IF a$="c" THEN GOTO 980
730 IF a$="p" THEN GOTO 1160
740 IF o=1 THEN cc=TEST(x,y):PLOT x,y,cc
:GOTO 770
750 PLOT x,y,pn
760 IF l=1 THEN DRAW x1,y1
770 GOTO 600
780 REM *****
790 REM * Draw box *
800 REM *****
810 LOCATE 1,1:PRINT c1$
820 LOCATE 1,1:PRINT "filled (y/n)"



```

```

830 a$=INKEY$
840 IF a$="y" THEN GOTO 910
850 IF a$="n" THEN GOTO 870
860 GOTO 830
870 PLOT x,y,pn:DRAW x,y1:DRAW x1,y1
880 DRAW x1,y:DRAW x,y
890 GOTO 530
900 REM
910 FOR i=x TO x1 STEP 2*SGN(x1-x)
920 PLOT i,y,pn:DRAW i,y1
930 NEXT i
940 GOTO 530
950 REM *****
960 REM * Circle *
970 REM *****
980 LOCATE 1,1:PRINT c1$
990 LOCATE 1,1:PRINT "x ratio , y ratio"
1000 INPUT xr,yr
1010 LOCATE 1,1:PRINT c1$
1020 LOCATE 1,1:PRINT "filled (y/n)"
1030 a$=INKEY$:IF a$="" THEN GOTO 1030
1040 IF a$="y" THEN f=1 ELSE f=0
1050 FOR i=1 TO 180
1060 x2=SIN(i)*xr
1070 y2=COS(i)*yr
1080 PLOT x+x2,y+y2,pn
1090 IF f=1 THEN DRAW x-x2,y+y2
1100 IF f=0 THEN PLOT x-x2,y+y2
1110 NEXT i
1120 GOTO 530
1130 REM *****
1140 REM * Change logical pen colour *
1150 REM *****
1160 LOCATE 1,1:PRINT c1$
1170 LOCATE 1,1:PRINT "change pen ";pn
1180 INPUT "to ";c1
1190 IF c1<0 OR c1>26 THEN GOTO 1160
1200 INK pn,c1
1210 GOTO 530

```


We can show you some of our efforts. The first is a nice landscape and the second is a surrealist computer workstation.



Chapter 4

GAMES OF SKILL

Wordslide



```
A L D G E
F C B N I
K H S R J
P Q X M X
U V W T O
```

For this game, you have to rearrange the letters into alphabetical order using the cursor control keys. The difficulty level controls the number of shuffles that the letter board goes through before you can start your work.

```
100 REM *****
110 REM * Wordslide *
120 REM *****
130 REM
140 MODE 0 : INK 0,0
150 PAPER 0 : BORDER 0
160 PEN 1 : INK 1,24
170 CLS : PEN 1
180 :
190 PRINT "      Wordslide"
200 PRINT "      -----"
```



```
210 PEN 2 : INK 2,26
220 LOCATE 1,5
230 PRINT "I will shuffle up "
240 PRINT:PRINT "the puzzle. All you"
250 PRINT:PRINT "have to do is use"
260 PRINT:PRINT "the cursor pad to"
270 PRINT:PRINT "try to rearrange the"
280 PRINT "order."
290 PEN 3 : INK 3,6,0 : SPEED INK 20,20
300 PRINT,,," Press any key"
310 PRINT,," to play"
320 IF INKEY$="" THEN GOTO 320
330 :
340 MODE 0 : INK 1,21 : PEN 1
350 REM
360 REM *****
370 REM * Set Up Board *
380 REM *****
390 REM
400 DIM a$(5,5) : DIM b$(5,5)
410 lis$="ABCDEFGHIJKLMNOPQRSTUVWXYZ*"
420 x = 0
430 FOR i = 1 TO 5
440   FOR j = 1 TO 5
450     x=x+1
460     a$(i,j)=MID$(lis$,x,1)
470     b$(i,j) = a$(i,j)
480   NEXT j
490 NEXT i
500 a$(5,5)=CHR$(15)+"1*"+CHR$(15)+"2"
510 b$(5,5)=a$(5,5)
520 :
530 GOSUB 1050
540 x = 5 : y = 5
550 LOCATE 1,20 : PEN 2
560 PRINT "Difficulty level"
570 INPUT "<1-99> ";le
580 IF le>0 AND le<100 THEN GOTO 610
```

```


590 LOCATE 9,21 : PRINT "          "
600 GOTO 550
610 CLS
620 REM
630 REM *****
640 REM * Scramble Board *
650 REM *****
660 REM
670 FOR i = 1 TO 1e+5
680   k=INT(4*RND(1))+1
690   SOUND 1,k*10,10,10
700   ON k GOSUB 1170,1260,1350,1440
710   a$(y,x)="1*2"
720 NEXT i
730 GOSUB 1050
740 :
750 k$=INKEY$ : IF k$="" THEN GOTO 750
760 SOUND 1,50,10,5
770 IF k$=CHR$(240) THEN GOSUB 1170
780 IF k$=CHR$(241) THEN GOSUB 1260
790 IF k$=CHR$(243) THEN GOSUB 1350
800 IF k$=CHR$(242) THEN GOSUB 1440
810 a$(y,x) = "1*2"
820 GOSUB 1050
830 REM
840 REM *****
850 REM * Done It??? *
860 REM *****
870 REM
880 FOR i = 1 TO 5
890   FOR j = 1 TO 5
900     IF a$(i,j) <> b$(i,j) THEN 750
910   NEXT j
920 NEXT i
930 :
940 FOR i = 0 TO 16
950   FOR j = 0 TO 4
960     INK 2,i : INK 0,j

```

```
970      SOUND 1,i*5+1,1,5
980    NEXT j
990  NEXT i
1000 :
1010 INK 0,0 : INK 1,5 : PEN 1
1020 PRINT:PRINT "      You did it"
1030 END
1040 :
1050 FOR i = 1 TO 5
1060   FOR j = 1 TO 5
1070     LOCATE j*2+5,i*2+2
1080     PRINT a$(i,j)
1090   NEXT j
1100 NEXT i
1110 RETURN
1120 REM
1130 REM *****
1140 REM * Swap 1 *
1150 REM *****
1160 REM
1170 IF y=1 THEN RETURN
1180 a$(y,x) = a$(y-1,x)
1190 y = y - 1
1200 RETURN
1210 REM
1220 REM *****
1230 REM * Swap 2 *
1240 REM *****
1250 REM
1260 IF y=5 THEN RETURN
1270 a$(y,x) = a$(y+1,x)
1280 y = y + 1
1290 RETURN
1300 REM
1310 REM *****
1320 REM * Swap 3 *
1330 REM *****
1340 REM
```

```
1350 IF x=5 THEN RETURN
1360 a$(y,x) = a$(y,x+1)
1370 x = x + 1
1380 RETURN
1390 REM
1400 REM *****
1410 REM * Swap 4 *
1420 REM *****
1430 REM
1440 IF x=1 THEN RETURN
1450 a$(y,x) = a$(y,x-1)
1460 x = x - 1
1470 RETURN
```

Reverse




You have to sort a string of nine digits into ascending numerical order. You can reverse any number of digits from 1 to 9 from the left-hand side. The machine will ask you how many you want.

```
100 REM *****
110 REM * Reverse *
120 REM *****
130 REM
140 MODE 0:INK 0,0:PEN 2
150 BORDER 0:PAPER 0
160 CLS:INK 1,24:INK 2,5
170 PRINT "      Reverse"
180 PRINT "      _____"
190 PRINT:PRINT
200 PEN 1
210 PRINT" I will display a "
```

```
220 PRINT"series of numbers"
230 PRINT"(1-9) which are in"
240 PRINT"the wrong order."
250 PRINT"You must get them"
260 PRINT"back into order."
270 PRINT "This is done by"
280 PRINT "REVERSING parts"
290 PRINT "of the list."
300 PEN 2
310 PRINT:PRINT"Press a key to play"
320 IF INKEY$=""THEN GOTO 320
330 MODE 0
340 DIM lis(9),temp(9)
350 REM
360 REM *****
370 REM * Set Up Board *
380 REM *****
390 REM
400 FOR i = 1 TO 9
410 place = INT(9*RND(1))+1
420 IF lis(place)<>0 THEN GOTO 410
430 lis(place)=i
440 NEXT i
450 GOSUB 780
460 LOCATE 1,5 : PEN 2
470 PRINT STRING$(20," ")
480 LOCATE 1,5
490 INPUT "No. to reverse ";n
500 IF n<1 OR n>9 THEN GOTO 460
510 REM
520 REM *****
530 REM * Reverse Number *
540 REM *****
550 REM
560 FOR i = 1 TO n
570 temp(i)=lis(i)
580 NEXT i
590 FOR i = 1 TO n
```

```
600 lis(i)=temp(n+1-i)
610 NEXT i
620 GOSUB 780
630 REM
640 REM *****
650 REM * Check for win *
660 REM *****
670 REM
680 FOR i = 1 TO 9
690 IF lis(i) <> i THEN GOTO 460
700 NEXT i
710 PRINT:PRINT:PRINT
720 PRINT "You got it!!"           ":END
730 REM
740 REM *****
750 REM * Display *
760 REM *****
770 REM
780 PEN 1
790 FOR i = 1 TO 9
800 LOCATE i*2,1
810 PRINT lis(i);
820 NEXT i
830 PRINT
840 RETURN
```

Flasher



With the Flasher, you have to repeat a sequence of flashing colours. You carry on following the Flasher until eventually he beats you into the ground. There are four possible flashing arrows corresponding to the cursor control keys. You have to repeat the sequence using these keys. Our record is a sequence of 17 (without using pen and paper). If you are a genius, the machine will take you up to 255 – and the best of luck.

```
100 REM *****
110 REM * Flasher *
120 REM *****
130 REM
140 record = 17
150 MODE 0
160 INK 0,0 : BORDER 0
170 INK 1,24 : PEN 1
180 :
```



```

190 PRINT "      Flasher"
200 PRINT "      -----"
210 LOCATE 1,4
220 INK 2,20 : PEN 2
230 PRINT "Press the cursor key"
240 PRINT "of the arrow that is"
250 PRINT "lit by the computer."
260 LOCATE 1,15 : INK 3,6 : PEN 3
270 PRINT "  Press any key"
280 PRINT
290 PRINT "      to play"
300 IF INKEY$="" THEN GOTO 300
310 PAPER 0 : CLS
320 INK 1,6:INK 2,2:INK 3,24:INK 4,18
330 REM
340 REM *****
350 REM * Draw Arrows *
360 REM *****
370 REM
380 FOR i = 1 TO 10
390 PLOT 100,200+i,1 : DRAW 100,0
400 PLOT 90,206 : DRAW 110,184+i*4
410 PLOT 540,200+i,2 : DRAW -100,0
420 PLOT 550,206 : DRAW 530,184+i*4
430 PLOT 320+i,390,3 : DRAW 0,-100
440 PLOT 326,400 : DRAW 305+i*4,380
450 PLOT 320+i,20,4 : DRAW 0,100
460 PLOT 320+i,0 : DRAW 305+i*4,20
470 NEXT i
480 REM
490 REM *****
500 REM * Display Series *
510 REM *****
520 REM
530 FOR i = 1 TO 10: a$=INKEY$:NEXT i
540 lis$="1"
550 INK 1,0:INK 2,0:INK 3,0:INK 4,0
560 :
```

```

570 lis$ = lis$ + RIGHT$( STR$( INT( 4*
    RND(1))+1),1)
580 PEN 12
590 FOR i = 1 TO LEN(lis$)
600   a$=MID$(lis$,i,1)
610   IF a$<>"1" THEN GOTO 670
620     INK 1,6
630     SOUND 1,10,10,15
640     GOSUB 850
650     INK 1,0
660     GOTO 830
670   IF a$<>"2" THEN GOTO 730
680     INK 2,2
690     SOUND 1,20,10,15
700     GOSUB 850
710     INK 2,0
720     GOTO 830
730   IF a$<>"3" THEN GOTO 790
740     INK 3,24
750     SOUND 1,30,10,15
760     GOSUB 850
770     INK 3,0
780     GOTO 830
790     INK 4,18
800     SOUND 1,40,10,15
810     GOSUB 850
820     INK 4,0
830 NEXT i
840 GOTO 910
850 FOR j=1 TO 300 : NEXT j : RETURN
860 REM
870 REM *****
880 REM * Input Players Guess *
890 REM *****
900 REM
910 FOR i = 1 TO LEN(lis$)
920   a$=MID$(lis$,i,1)
930   k$=INKEY$ : IF k$="" THEN 930

```


```

940   IF k$<CHR$(240) OR k$>CHR$(243)
      THEN 930
950   IF k$=CHR$(242) THEN INK 1, 6 :
      SOUND 1,10,10,15
960   IF k$=CHR$(243) THEN INK 2, 2 :
      SOUND 1,20,10,15
970   IF k$=CHR$(240) THEN INK 3,24 :
      SOUND 1,30,10,15
980   IF k$=CHR$(241) THEN INK 4,18 :
      SOUND 1,40,10,15
990   FOR k=1 TO 100 : NEXT k
1000  IF k$=CHR$(242)AND a$="1" THEN
      GOTO 1050
1010  IF k$=CHR$(243)AND a$="2" THEN
      GOTO 1050
1020  IF k$=CHR$(240)AND a$="3" THEN
      GOTO 1050
1030  IF k$=CHR$(241)AND a$="4" THEN
      GOTO 1050
1040  GOTO 1290
1050  FOR j=1 TO 4
1060    INK j,0
1070  NEXT j
1080  NEXT i
1090  REM
1100  REM *****
1110  REM * Series Correct *
1120  REM *****
1130  REM
1140  FOR j = 1 TO 5
1150    FOR i = 1 TO 4
1160      INK i,INT(27*RND(1))+1
1170      FOR k=1 TO 20 : NEXT k
1180      INK i,0
1190    NEXT i
1200    SOUND 1,j*10,10,15
1210  NEXT j
1220  FOR i = 1 TO 200 : NEXT i

```

```
1230 GOTO 570
1240 REM
1250 REM *****
1260 REM * Game Over *
1270 REM *****
1280 REM
1290 FOR i = 1 TO 10
1300   SOUND 1,10*VAL(a$),10,15
1310   INK VAL(a$),26
1320   FOR j=1 TO 20
1330     INK 0,INT(27*RND(1))+1
1340   NEXT j
1350   INK VAL(a$),0
1360 NEXT i
1370 length = LEN(lis$)
1380 INK 0,0:PAPER 0:MODE 1
1390 INK 1,24
1400 PEN 1: PRINT "You Scored ";length
1410 IF record > length THEN 1450
1420 PRINT:PRINT"A new Record!!!"
1430 record = length
1440 GOTO 1460
1450 PRINT:PRINT"The Record stands at ";
record
1460 PRINT:PRINT:PRINT:PRINT
1470 LOCATE 1,20:INPUT "Another go ";g$
1480 IF g$="" THEN 1470
1490 IF g$="y" THEN GOTO 150
1500 IF g$<>"n" THEN GOTO 1470
```

Fruit Machine



You have a three-window fruit machine with “random hold after win” and six nudges (upwards). You “pull the arm” with the space bar. If nudges show, you must use the space bar to pick up as many nudges as possible. Then you nudge the reels with 1, 2 or 3.

```
100 REM *****
110 REM * Fruit machine *
120 REM *****
130 ENV 1,5,5,2
140 MODE 1
150 INK 1,2:INK 2,6
160 INK 3,8:INK 4,18
170 INK 5,20:INK 6,24
180 INK 7,26
190 INK 0,0
200 INK 8,15:INK 9,7
```

```
210 INK 10,17
220 PEN 6
230 PEN 3
240 LOCATE 13,1:PRINT "Fruit machine"
250 PEN 1
260 PRINT
270 PRINT " Press space bar to start ree
1s"
280 PRINT
290 PEN 2
300 PRINT "Use numbers <1> <2> and <3>"
310 PRINT "for hold and nudge"
320 PRINT
330 PEN 3
340 PRINT "Press any key to get nudge nu
mber"
350 PRINT
360 PRINT
370 PEN 2
380 PRINT "Press any key to start"
390 IF INKEY$="" THEN 390
400 MODE 0
410 WINDOW #2,5,1,1,12
420 WINDOW #3,11,7,1,12
430 WINDOW #4,17,13,1,12
440 RESTORE 1610
450 money=200
460 DIM a(30),b(30),c(30)
470 FOR i=1 TO 30
480 READ a(i)
490 NEXT i
500 FOR i=1 TO 30
510 READ b(i)
520 NEXT i
530 FOR i=1 TO 30
540 READ c(i)
550 NEXT i
560 FOR i=2 TO 4
```

```
570 PAPER #i,15
580 PEN #i,1
590 CLS #i
600 NEXT i
610 chan=3
620 a1=1
630 b1=1
640 c1=1
650 t1=INT(5*RND(TIME)+4)
660 t2=INT(10*RND(TIME)+t1)
670 t3=INT(10*RND(TIME)+t2)
680 FOR j=1 TO t3
690 IF h1=1 THEN 760
700 t1=t1-1
710 IF t1<0 THEN 760
720 a1=a1+1
730 IF a1=31 THEN a1=1
740 chan=2:l=a(a1):GOSUB 1700
750 SOUND 1,15,10,15,1
760 IF h2=1 THEN GOTO 830
770 t2=t2-1
780 IF t2<0 THEN GOTO 830
790 b1=b1+1
800 IF b1=31 THEN b1=1
810 chan=3:l=b(b1):GOSUB 1700
820 SOUND 1,32,10,15,1
830 IF h3=1 THEN 900
840 t3=t3-1
850 IF t3<0 THEN 900
860 c1=c1+1
870 IF c1=31 THEN c1=1
880 chan=4:l=c(c1):GOSUB 1700
890 SOUND 1,45,10,15,1
900 REM
910 NEXT j
920 IF INT(10*RND(TIME)+1)=10 THEN GOSUB
2520
930 GOSUB 2210
```

```
940 IF win <>0 THEN GOSUB 2420
950 money=money+win
960 GOSUB 2340
970 GOSUB 2050
980 IF INT(10*RND(TIME)+1)>8 THEN GOSUB
1830
990 LOCATE 1,16
1000 PEN 10
1010 PAPER 0
1020 PRINT "Press a key to start";
1030 IF INKEY$="" THEN 1030
1040 LOCATE 1,16
1050 PRINT SPACE$(20);
1060 GOTO 650
1070 REM *****
1080 REM * orange *
1090 REM *****
1100 PAPER #chan,8
1110 PEN #chan,0
1120 PRINT #chan,"O R A";
1130 PRINT #chan," N G ";
1140 PRINT #chan," E ";
1150 RETURN
1160 REM *****
1170 REM * lemon *
1180 REM *****
1190 PAPER #chan,6
1200 PEN #chan,0
1210 PRINT #chan,"LE ";
1220 PRINT #chan," M ";
1230 PRINT #chan," ON";
1240 RETURN
1250 REM *****
1260 REM * bar *
1270 PAPER #chan,1
1280 REM *****
1290 PEN #chan,7
1300 PRINT #chan,"-----";
```



```

1310 PRINT #chan," BAR ";
1320 PRINT #chan,"-----";
1330 RETURN
1340 REM *****
1350 REM * cherry *
1360 REM *****
1370 PAPER #chan,4
1380 PEN #chan,2
1390 PRINT #chan,"C H E";
1400 PRINT #chan,"      ";
1410 PRINT #chan,"R R Y";
1420 RETURN
1430 REM *****
1440 REM * strawberry *
1450 REM *****
1460 PAPER #chan,2
1470 PEN #chan,4
1480 PRINT #chan,"STRAW";
1490 PRINT #chan,"      ";
1500 PRINT #chan,"BERRY";
1510 RETURN
1520 REM *****
1530 REM * grape *
1540 REM *****
1550 PAPER #chan,3
1560 PEN #chan,9
1570 PRINT #chan,"      ";
1580 PRINT #chan,"GRAPE";
1590 PRINT #chan,"      ";
1600 RETURN
1610 DATA 3,3,4,2,3,5,4,2,1,5,4,1,3,5,2
1620 DATA 5,2,4,3,5,6,5,1,3,2,4,5,2,4,2
1630 DATA 3,4,5,1,1,3,4,3,1,3,4,2,4,6,1
1640 DATA 3,5,3,4,2,5,1,3,2,5,4,2,3,4,1
1650 DATA 3,2,4,1,1,4,5,2,5,3,6,3,2,1,3
1660 DATA 4,5,1,2,3,4,3,5,3,5,1,5,2,5,1
1670 REM *****
1680 REM * Choose reel *
1690 REM *****

```


```
1700 IF 1=1 THEN GOSUB 1370
1710 IF 1=2 THEN GOSUB 1100
1720 IF 1=3 THEN GOSUB 1190
1730 IF 1=4 THEN GOSUB 1460
1740 IF 1=5 THEN GOSUB 1550
1750 IF 1=6 THEN GOSUB 1270
1760 PEN #chan,7
1770 PAPER #chan,0
1780 PRINT #chan,"=====";
1790 RETURN
1800 REM *****
1810 REM * Hold *
1820 REM *****
1830 PAPER 0
1840 FOR j=1 TO 24
1850 FOR i=0 TO 3
1860 LOCATE 8,16
1870 INK 15,i+j
1880 PEN 15:PRINT "Hold"
1890 SOUND 1,j*i,2,15
1900 NEXT i
1910 NEXT j
1920 a$=INKEY$
1930 PAPER 5
1940 PEN 14
1950 IF a$="1" THEN h1=1:LOCATE 1,13:PRI
NT "Held"
1960 IF a$="2" THEN h2=1:LOCATE 7,13:PRI
NT "Held"
1970 IF a$="3" THEN h3=1:LOCATE 13,13:PR
INT "Held"
1980 IF a$="c" THEN GOSUB 2050
1990 IF a$=" " THEN 2010
2000 GOTO 1920
2010 RETURN
2020 REM *****
2030 REM * Cancel hold *
2040 REM *****
2050 PAPER 1
```

```
2060 PEN 6
2070 LOCATE 1,13
2080 PRINT "Hold"
2090 LOCATE 7,13
2100 PRINT "Hold"
2110 LOCATE 13,13
2120 PRINT "Hold"
2130 h1=0
2140 h2=0
2150 h3=0
2160 PEN 0
2170 PAPER 0
2180 LOCATE 8,16:PRINT "      "
2190 LOCATE 1,15:PRINT SPACE$(20);
2200 RETURN
2210 win =0
2220 aa=a(a1)
2230 bb=b(b1)
2240 cc=c(c1)
2250 IF aa=1 AND bb=1 AND cc=1 THEN win=
20:GOTO 2300
2260 IF aa=6 AND bb=6 AND cc=6 THEN win=
1000:GOTO 2300
2270 IF aa=6 AND bb=6 THEN win=500:GOTO
2300
2280 IF aa=bb AND bb=cc AND cc=aa THEN w
in=20*a+10:GOTO 2300
2290 IF aa=bb THEN win=10*aa
2300 RETURN
2310 REM *****
2320 REM * Cash *
2330 REM *****
2340 LOCATE 1,24
2350 PEN 12
2360 PAPER 0
2370 PRINT "cash=";money;" "
2380 RETURN
```

```
2390 REM *****
2400 REM * Display win *
2410 REM *****
2420 PAPER 0
2430 PEN 11
2440 LOCATE 5,21
2450 PRINT "You Won ";win
2460 FOR j=1 TO 100
2470 SOUND 1,j*2.2,1,15
2480 NEXT j
2490 LOCATE 5,21
2500 PRINT SPACE$(13)
2510 RETURN
2520 LOCATE 8,16
2530 PAPER 0
2540 PEN 2
2550 PRINT "Nudge"
2560 FOR j=0 TO 6
2570 LOCATE j*2+3,15
2580 SOUND 1,j*10,2,15
2590 PEN 5
2600 PRINT j
2610 IF INKEY$<>" " THEN GOTO 2670
2620 LOCATE j*2+3,15
2630 PEN 1
2640 PRINT j
2650 NEXT j
2660 GOTO 2560
2670 LOCATE 8,16
2680 PRINT "      "
2690 nd=j
2700 IF nd=0 THEN GOTO 2950
2710 a$=INKEY$
2720 IF a$="" THEN 2710
2730 IF a$<>"1" THEN GOTO 2770
2740 a1=a1+1
2750 IF a1=31 THEN a1=1
2760 chan=2:1=a(a1):GOSUB 1700
```

```
2770 IF a$<>"2" THEN GOTO 2810
2780 b1=b1+1
2790 IF b1=31 THEN b1=1
2800 chan=3:l=b(b1):GOSUB 1700
2810 IF a$<>"3" THEN 2850
2820 c1=c1+1
2830 IF c1=31 THEN c1=1
2840 chan=4:l=c(c1):GOSUB 1700
2850 LOCATE nd*2+3,15
2860 PEN 1
2870 PRINT nd
2880 nd=nd-1
2890 IF nd=0 THEN 2950
2900 GOSUB 2210:IF win<>0 THEN GOTO 940
2910 LOCATE nd*2+3,15
2920 PEN 5
2930 PRINT nd
2940 GOTO 2710
2950 RETURN
```

Pennies



This is the old three cups and a coin. The difficulty level controls the number of shuffles of the cups. As the difficulty level increases, so does the number of shuffles - but so does the speed of the shuffling. You then have to tell the machine where the penny is by entering 1, 2 or 3.

```
100 REM *****
110 REM * Pennies *
120 REM *****
130 REM
140 MODE 0
150 INK 0,0
160 PAPER 0
170 CLS
180 BORDER 0
190 INK 1,24 : PEN 1
200 PRINT "      Pennies"
```

```

210 PRINT "      -----"
220 PRINT:PRINT:PRINT
230 INK 2,26 : PEN 2
240 PRINT "  I have three cups"
250 PRINT "under one of them is";
260 PRINT "a pea. You must try"
270 PRINT "to follow the pea as";
280 PRINT "I move it around."
290 INK 3,21,6:PEN 3 : PRINT
300 PRINT "      SOUNDS EASY!"
310 PRINT "      BUT IT'S NOT":PRINT
320 PEN 2 :PRINT "Press a key to play"
330 IF INKEY$="" THEN GOTO 330
340 CLS : INK 1,21 : PEN 1
350 LOCATE 3,2:INPUT "Input level 1-9";l
360 CLS : PEN 1
370 LOCATE 6,5 : PRINT "o"
380 LOCATE 9,5 : PRINT "o"
390 LOCATE 12,5 : PRINT "o"
400 DIM a(3)
410 pea = INT(3*RND(1))+1
420 a(pea) = 1
430 LOCATE 1,1
440 INK 2,6 : PEN 2
450 PRINT "  The pea starts"
460 PRINT "  under pod number";pea
470 PEN 1
480 REM
490 REM *****
500 REM * Shuffle Pennies *
510 REM *****
520 REM
530 FOR i = 1 TO INT(1*3*RND(1))+1
540 which = INT(3*RND(1))+1
550 ON which GOSUB 810,960,1110
560 NEXT i
570 REM

```

```


580 REM *****
590 REM * Choose a cup *
600 REM *****
610 REM
620 LOCATE 1,15
630 INPUT "Which one ";b$
640 IF b$="" THEN GOTO 630
650 IF b$<"1" OR b$>"3" THEN GOTO 630
660 IF a(VAL(b$)) = 0 THEN 690
670 PRINT "You got it!!!"
680 END
690 PRINT "You missed it"
700 FOR i = 1 TO 3
710 IF a(i)=1 THEN GOTO 730
720 NEXT i
730 PRINT "It was under cup"
740 PRINT "number ";i
750 END
760 REM
770 REM *****
780 REM * Move 1 *
790 REM *****
800 REM
810 a(0)=a(1):a(1)=a(2):a(2)=a(0)
820 LOCATE 6,5 : PRINT "   "
830 FOR loop = 0 TO 3
840   LOCATE 6+loop,4 : PRINT"o"
850   LOCATE 9-loop,6 : PRINT"o"
860   LOCATE 6+loop,4 : PRINT" "
870   LOCATE 9-loop,6 : PRINT" "
880 NEXT loop
890 LOCATE 6,5 : PRINT "o o"
900 RETURN
910 REM
920 REM *****
930 REM * Move 2 *
940 REM *****
950 REM

```



```
960 a(0)=a(1):a(1)=a(3):a(3)=a(0)
970 LOCATE 6,5 : PRINT " o  "
980 FOR loop = 0 TO 6
990   LOCATE 6+loop,4 : PRINT"o"
1000   LOCATE 12-loop,6 : PRINT"o"
1010   LOCATE 6+loop,4 : PRINT" "
1020   LOCATE 12-loop,6 : PRINT" "
1030 NEXT loop
1040 LOCATE 6,5 : PRINT"o o o"
1050 RETURN
1060 REM
1070 REM *****
1080 REM * Move 3 *
1090 REM *****
1100 REM
1110 a(0)=a(2):a(2)=a(3):a(3)=a(0)
1120 LOCATE 9,5 : PRINT "   "
1130 FOR loop = 0 TO 3
1140   LOCATE 9+loop,4 : PRINT"o"
1150   LOCATE 12-loop,6 : PRINT"o"
1160   LOCATE 9+loop,4 : PRINT" "
1170   LOCATE 12-loop,6 : PRINT" "
1180 NEXT loop
1190 LOCATE 9,5 : PRINT"o o"
1200 RETURN
```

Leapfrog



You have two groups of frogs. As you may realise, a frog can leap over another one. The object is to make the two groups of frogs change places. You enter the number of the frog that will leap.

```
100 REM *****
110 REM * Leap frog *
120 REM *****
130 MODE 1
140 PAPER 0
150 ENT 1,5,1,1,10,-1,1,10,1,1,5,-1,1
160 ENV 1,5,3,1,1,0,16,5,-3,2
170 INK 0,11
180 INK 1,24:INK 2,3
190 INK 3,18
200 PEN 3
210 LOCATE 16,1:PRINT "Leap frog"
```

```
220 PEN 1
230 PRINT
240 PRINT "Two groups of frog are trying
to"
250 PRINT "cross the pond on a line of"
260 PRINT "lily pads, Each frog can jump
"
270 PRINT "over another frog onto the sp
are"
280 PRINT "lily pad."
290 PEN 2:PRINT
300 PRINT "The objective of this game is
to "
310 PRINT "swap all the yellow frogs to
brown"
320 PRINT "frogs and vica verca."
330 PRINT
340 PEN 3
350 PRINT "Your score is how many moves
it takes"
360 PRINT "you to move all the frogs."
370 PRINT:PRINT " Press any key to sta
rt."
380 IF INKEY$="" THEN GOTO 380
390 CLS
400 REM *****
410 REM * U.D.G.s *
420 REM *****
430 SYMBOL AFTER 91
440 SYMBOL 91,0,2,13,31,126,252,242,113
450 SYMBOL 93,0,64,176,248,254,63,79,78
460 SYMBOL 94,126,223,227,126,0,0,0,0
470 DIM a(11)
480 FOR i=1 TO 5
490 a(i)=1:a(6+i)=2
500 NEXT i
510 PEN 3
520 FOR i=1 TO 11
```

```


530 LOCATE 1+i*3,11:PRINT CHR$(94)
540 r=110
550 NEXT i
560 REM *****
570 REM * Draw screen *
580 REM *****
590 DEG:FOR i=90 TO 181 STEP 0.36
600 y=COS(i+180)*r
610 x=SIN(i)*320
620 PLOT 320+x,400,1:DRAW 320+x,250+y
630 PLOT 320-x,400,1:DRAW 320-x,250+y
640 NEXT i
650 FOR i=1 TO 8
660 FOR x1=30 TO 630 STEP 529:y=270
670 x=x1
680 xd=RND(TIME)*3-1:xd=xd/2
690 yd=-2
700 FOR j=1 TO RND(TIME)*30+30
710 x=x+xd:y=y-yd
720 PLOT x,y,3
730 NEXT j
740 FOR j=1 TO 15
750 x=x+xd:y=y-yd
760 PLOT x-2,y,2:PLOT x,y,2:PLOT x+2,y,2
770 NEXT j
780 NEXT x1
790 NEXT i
800 PLOT x+xd,y+4,3
810 REM *****
820 REM * Main loop *
830 REM *****
840 FOR i=1 TO 11
850 LOCATE 1+i*3,10
860 PEN a(i)
870 IF a(i)=1 THEN PRINT "["
880 IF a(i)=2 THEN PRINT "]"
890 IF a(i)=0 THEN PRINT " "
900 LOCATE i*3,14:PRINT i

```

```
910 NEXT i
920 LOCATE 3,17:PRINT SPACE$(20)
930 LOCATE 3,17:PRINT "Frog i-";
940 INPUT f
950 IF f<1 OR f>11 OR f<>INT(f) THEN 920
960 IF a(f)=0 THEN GOSUB 1070:GOTO 840
970 FOR i=1 TO 11:IF a(i)=0 THEN s=i:GOT
0 990
980 NEXT i
990 IF ABS(s-f)>2 THEN 1050
1000 xx=a(s):a(s)=a(f):a(f)=xx
1010 GOSUB 1260
1020 PAPER 1:PEN 2
1030 m=m+1:LOCATE 14,2:PRINT "Moves =";m
1040 PAPER 0
1050 GOSUB 1160
1060 GOTO 840
1070 LOCATE 6,1
1080 PRINT "you cant make a space jump"
1090 FOR i=1 TO 1000:NEXT i
1100 LOCATE 6,1
1110 PRINT SPACE$(26)
1120 RETURN
1130 REM *****
1140 REM * Check for win *
1150 REM *****
1160 FOR i=1 TO 5
1170 IF a(i)=2 THEN NEXT i:GOTO 1190
1180 GOTO 1210
1190 FOR i=7 TO 11
1200 IF a(i)=1 THEN NEXT i:GOTO 1220
1210 RETURN
1220 GOTO 1420
1230 REM *****
1240 REM * Make frog jump *
1250 REM *****
1260 IF s-f<0 THEN ss=1:frog=93
1270 IF s-f>0 THEN ss=-1:frog=91
```

```
1280 PEN a(s)
1290 LOCATE 1+f*3,10:PRINT " "
1300 LOCATE 1+f*3,9:PRINT CHR$(frog)
1310 FOR i=f*3 TO s*3 STEP -ss
1320 SOUND 1,(s*3-i)+200,15,15,1,1,14
1330 LOCATE 1+i,9:PRINT CHR$(frog)
1340 FOR j=1 TO 100:NEXT j
1350 LOCATE i+1,9:PRINT " "
1360 NEXT i
1370 LOCATE i-2*(ss=1),10:PRINT CHR$(frog)
1380 RETURN
1390 REM *****
1400 REM * Game over *
1410 REM *****
1420 PAPER 1:PEN 2
1430 LOCATE 8,2:PRINT "You did it in ";m
; " moves"
1440 LOCATE 15,3:PRINT "Play again?"
1450 a$=INKEY$
1460 IF a$="" THEN 1450
1470 IF a$="y" THEN RUN
1480 IF a$="n" THEN END
1490 GOTO 1450
```

Twister



This is a multicoloured shuffle cylinder based on an old Hungarian puzzle. Full instructions are given in the listing.

```
100 REM *****
110 REM * Twister *
120 REM *****
130 REM
140 MODE 0:INK 0,0
150 BORDER 0:PAPER 0
160 CLS:INK 1,24
170 PRINT " Twister"
180 PRINT " ~~~~~"
190 PRINT:PRINT:PRINT
200 PRINT " There is no end to "
210 PRINT "this puzzle. Just a "
220 PRINT "long time taken not "
```

```

230 PRINT "getting there???"      "
240 PRINT " Use number pad to    "
250 PRINT "move the reels.      "
260 PRINT:PRINT "Press a key to play"
270 IF INKEY$="" THEN GOTO 270
280 CLS
290 REM
300 REM *****
310 REM * Randomize Barrel *
320 REM *****
330 REM
340 DIM barrel(3,6)
350 FOR i = 1 TO 3
360   FOR j = 1 TO 6
370     p1=INT(3*RND(1))+1
380     p2=INT(6*RND(1))+1
390     IF barrel(p1,p2) <> 0 THEN 370
400     barrel(p1,p2)=j
410   NEXT j
420 NEXT i
430 barrel(3,6)=0
440 x=6 : y=3
450 REM
460 REM *****
470 REM * Set Up Colours For Reels *
480 REM *****
490 REM
500 INK 0,0 : INK 1,24 : INK 2,5
510 INK 3,6 : INK 4,26 : INK 5,18
520 INK 6,15
530 REM
540 REM *****
550 REM * Display Barrel *
560 REM *****
570 REM
580 PEN 6 : PRINT"      Twister"
590 FOR i = 1 TO 3
600   FOR j = 1 TO 6


```



```
610     LOCATE j*2+3,i+5*2
620     PEN barrel(i,j)
630     PRINT CHR$(143);
640     NEXT j
650 NEXT i
660 REM
670 REM *****
680 REM * Read Keyboard *
690 REM *****
700 REM
710 a$=INKEY$ : IF a$="" THEN GOTO 710
720 IF a$<"1" OR a$>"9" THEN GOTO 710
730 ON VAL(a$) GOSUB 760,1110,750,770,1050,
    780,790,1210,800
740 GOTO 590
750 reel = 3 :GOSUB 980:RETURN
760 reel = 3 :GOSUB 860:RETURN
770 reel = 2 :GOSUB 860:RETURN
780 reel = 2 :GOSUB 980:RETURN
790 reel = 1 :GOSUB 860:RETURN
800 reel = 1 :GOSUB 980:RETURN
810 REM
820 REM *****
830 REM * Revolve Left *
840 REM *****
850 REM
860 barrel(reel,0)=barrel(reel,1)
870 FOR i = 1 TO 5
880     barrel(reel,i)=barrel(reel,i+1)
890 NEXT i
900 barrel(reel,6)=barrel(reel,0)
910 IF reel = y THEN x=x-1 : IF x=0
    THEN x=6
920 RETURN
930 REM
940 REM *****
950 REM * Revolve Right *
960 REM *****
```

```
970 REM
980 barrel(reel,0)=barrel(reel,6)
990 FOR i = 6 TO 2 STEP -1
1000   barrel(reel,i)=barrel(reel,i-1)
1010 NEXT i
1020 barrel(reel,1)=barrel(reel,0)
1030 IF reel = y THEN x=x+1 : IF x=7
      THEN x=1
1040 RETURN
1050 RETURN : REM * Key 5 Pressed *
1060 REM
1070 REM *****
1080 REM * Drop a piece down *
1090 REM *****
1100 REM
1110 IF y=1 THEN RETURN
1120 barrel(y,x)=barrel(y-1,x)
1130 barrel(y-1,x)=0
1140 y=y-1
1150 RETURN
1160 REM
1170 REM *****
1180 REM * Move a piece up *
1190 REM *****
1200 REM
1210 IF y=3 THEN RETURN
1220 barrel(y,x)=barrel(y+1,x)
1230 barrel(y+1,x)=0
1240 y=y+1
1250 RETURN
```

Mummies' Revenge



You have crash-landed on a remote island whose inhabitants have long since died. BUT their long-dead spirits have arisen to chase you away. To avoid them you must force them into the lava pits. The cursor controls aid your movement, but you are very weak so must hurry before you collapse.

```
100 REM *****
110 REM * Mummies' revenge *
120 REM *****
130 MODE 1
140 SYMBOL AFTER 91
150 SYMBOL 92,24,24,31,25,24,24,24,28
160 SYMBOL 91,24,24,248,152,24,24,24,56
170 SYMBOL 93,1,7,63,15,32,127,63,3
180 INK 0,0
190 INK 1,5
```

```

200 INK 2,8
210 INK 3,6,24
220 PAPER 0
230 BORDER 0
240 SPEED INK 1,1
250 p=1
260 REM *****
270 REM * Instructions *
280 REM *****
290 PEN 3
300 a$="Mummies' revenge"
310 GOSUB 1860
320 a$="":GOSUB 1860
330 PEN 1
340 a$="You are trapped!! .The only way
of"
350 GOSUB 1860
360 a$="getting off the island is to lur
e"
370 GOSUB 1860
380 a$="the Mummies into the pits that
are"
390 GOSUB 1860
400 a$="dotted around the place,Or to le
t"
410 GOSUB 1860
420 a$="themselves bump into each other.
"
430 GOSUB 1860
440 a$="The controls for moving your"
450 GOSUB 1860
460 a$="man are as follows..."
470 GOSUB 1860
480 PAPER 2: PEN 1
490 a$="      up      ":GOSUB 1860
500 a$="      "+CHR$(244)+"      ":GOSU
B 1860
510 a$="      |      ":GOSUB 1860

```

```

520 a$="left "+CHR$(247)+"- "+CHR$(248)+
" -"+CHR$(246)+" right"
530 GOSUB 1860
540 a$="          |          ":GOSUB 1860
550 a$="          "+CHR$(245)+"          ":GOSU
B 1860
560 a$="    down          ":GOSUB 1860
570 PAPER 0
580 a$="If you walk into a pit or"
590 GOSUB 1860
600 a$="are caught by a zombie then you"
610 GOSUB 1860
620 a$="DIE!!!,and the game is over"
630 GOSUB 1860
640 a$=CHR$(231)+" hole "+CHR$(92)+"
zombie"
650 GOSUB 1860
660 PEN 3:LOCATE 12,20:PRINT CHR$(231)
670 PEN 1
680 a$="":GOSUB 1860
690 a$=CHR$(248)+" man "+CHR$(93)+"
sea  "
700 GOSUB 1860
710 PEN 2:p=24
720 a$="Press any key to land on the isl
and"
730 GOSUB 1860
740 IF INKEY$="" THEN GOTO 740
750 DIM x(15),y(15),x1(15),y1(15),d(15)
760 MODE 1
770 REM *****
780 REM * Set everything up *
790 REM *****
800 INK 0,9
810 INK 1,20
820 INK 2,24,6
830 INK 3,16
840 BORDER 0

```

```

850 SPEED INK 1,1
860 zombie1$=CHR$(15)+"1"+CHR$(92)
870 zombie2$=CHR$(15)+"1"+CHR$(91)
880 hole$   =CHR$(15)+"2"+CHR$(231)
890 me$     =CHR$(15)+"3"+CHR$(248)
900 X=20
910 Y=12
920 LOCATE X,Y:PRINT ME$
930 FOR i=1 TO 15
940 x(i)=INT(32*RND(TIME)+4)
950 y(i)=INT(20*RND(TIME)+3)
960 LOCATE x(i),y(i)
970 PRINT zombie$;
980 d(i)=0
990 NEXT i
1000 FOR i=0 TO 640 STEP 4
1010 PLOT i,0,1:DRAW i,INT(16*RND(TIME)+
1)
1020 PLOT i,399:DRAW i,399-INT(16*RND(TI
ME)+1)
1030 NEXT i
1040 FOR i=0 TO 399 STEP 4
1050 PLOT 0,i:DRAW INT(16*RND(TIME)+1),i
1060 PLOT 640,i:DRAW 640-INT(16*RND(TIME
)+1),i
1070 NEXT i
1080 REM *****
1090 REM * Main loop *
1100 REM *****
1110 FOR i=1 TO 15
1120 xx=INT(30*RND(TIME)+5)
1130 yy=INT(20*RND(TIME)+3)
1140 IF TEST(16*xx-8,408-(16*yy))<>0 THE
N GOTO 1120
1150 LOCATE xx,yy:PRINT hole$
1160 NEXT i
1170 TM=0
1180 FOR i=1 TO 15

```

```

1190 tm=tm+1
1200 IF d(i)=1 THEN GOTO 1500
1210 x1(i)=x(i):y1(i)=y(i)
1220 IF x(i)<x THEN x(i)=x(i)+1:z$=zombi
el$
1230 IF x(i)>x THEN x(i)=x(i)-1:z$=zombi
er$
1240 IF y(i)>y THEN y(i)=y(i)-1
1250 IF y(i)<y THEN y(i)=y(i)+1
1260 tx=16*x(i)-8:ty=408-(16*y(i))
1270 REM *****
1280 REM * Mummy hit another mummy *
1290 REM *****
1300 IF TEST(tx,ty)<>1 THEN 1360
1310 d(i)=1:d=d+1:LOCATE x(i),y(i)
1320 PRINT " ";
1330 LOCATE x1(i),y1(i):PRINT " ";
1340 GOTO 1450
1350 REM *****
1360 REM * Hit a pit *
1370 REM *****
1380 IF TEST(tx,ty)<>2 THEN 1420
1390 d(i)=1:d=d+1:LOCATE x(i),y(i):PRINT
" ";
1400 LOCATE x1(i),y1(i):PRINT " ";
1410 GOTO 1450
1420 LOCATE x(i),y(i)
1430 PRINT z$;
1440 LOCATE x1(i),y1(i)
1450 PRINT " "
1460 IF d=15 THEN GOTO 1720
1470 REM *****
1480 REM * Move player *
1490 REM *****
1500 a$=INKEY$
1510 IF a$="" THEN GOTO 1630
1520 xx=x:yy=y
1530 IF a$=CHR$(240) AND y>2 THEN y=y-1
1540 IF a$=CHR$(241) AND y<24 THEN y=y+1

```

```

1550 IF a$=CHR$(242) AND x>2 THEN x=x-1
1560 IF a$=CHR$(243) AND x<38 THEN x=x+1
1570 LOCATE xx,yy
1580 PRINT " "
1590 tx=x*16-8:ty=408-(y*16)
1600 IF TEST(tx,ty)=2 THEN GOTO 1700
1610 LOCATE x,y
1620 PRINT me$;
1630 NEXT i
1640 GOTO 1180
1650 REM *****
1660 REM * Game over *
1670 REM *****
1680 GOSUB 1930:PEN 2:a$="You died":p=12
:GOSUB 1860
1690 GOTO 1770
1700 GOSUB 1930:p=12:a$="You fell into a
pit":GOSUB 1860
1710 GOTO 1770
1720 GOSUB 1930:p=12:a$="you got all the
Mummies "
1730 GOSUB 1860
1740 a$="50 bonus time units lost"
1750 GOSUB 1860
1760 tm=tm-50
1770 p=14:a$="Game Over":GOSUB 1860
1780 sc=TM:a$="Score = "+STR$(sc):GOSUB
1860
1790 p=p+2:PEN 1:INK 1,6
1800 a$="Do you want to play again?"
1810 GOSUB 1860
1820 a$=INKEY$
1830 IF a$="y" THEN RUN
1840 IF a$<>"n" THEN GOTO 1820
1850 END
1860 FOR i=1 TO LEN (a$)
1870 LOCATE 20-LEN(a$)/2+i,p:PRINT MID$(
a$,i,1)
1880 SOUND 1,ASC(MID$(a$,i,1)),1,15

```




```
1890 FOR j=1 TO 3:NEXT j
1900 NEXT i
1910 p=p+1
1920 RETURN
1930 CLS:PAPER 0
1940 INK 0,0
1950 RETURN
```

Chapter 5

“WEAPON” GAMES

Leveller



Your plane is running out of fuel and you must land. In order to do so, you must knock down every building in sight to prepare your glide path. Watch it though, some of the buildings are pretty high. If you do land, your next flight out will be tougher because the next city has even taller buildings.

```
100 REM *****
110 REM * Leveller *
120 REM *****
130 MODE 1
140 PEN 1:INK 1,5
141 INK 0,0:PAPER 0
150 LOCATE 16,1:PRINT "Leveller"
155 PRINT
160 PEN 2
170 PRINT "Your aircraft is running out
of fuel"
```

```

180 PRINT "So you must find a place to l
and.
185 PRINT
190 PRINT "The only place your fuel supp
ly "
200 PRINT "will get you to is an old cit
y"
205 PRINT
210 PRINT "To land you must destroy all
the"
220 PRINT "buildings.Use the space-bar t
o drop "
230 PRINT "your bombs. Be quick as there
is"
240 PRINT "little time."
244 PRINT
245 PEN 3:PRINT "          press any key to
start"
250 IF INKEY$="" THEN GOTO 250
260 LV=4:SC=0
270 ENT 1,3,4,60:ENV 1,4,-3,80
280 SYMBOL AFTER 90
290 SYMBOL 91,0,0,128,192,254,255,254,0
300 SYMBOL 92,0,24,24,24,60,60,126,126
310 SYMBOL 93,255,153,153,255,255,153,15
3,255
320 INK 0,20:INK 1,13:INK 2,21:INK 3,5
330 B$=CHR$(252)
340 MODE 1
350 bx=0:by=0
360 GOSUB 740
370 REM *****
380 REM * MAIN LOOP FOR PLANE *
390 REM *****
400 FOR Y=2 TO 22:FOR X=2 TO 37
410 PEN 3:LOCATE X,Y:PRINT " [";
420 IF TEST((16*X)+1,(16*(25-Y))-1)=1 TH
EN 540

```

```

430 IF INKEY$=" " AND F=0 THEN F=1:BX=X+
1:BY=Y+1:IF BY>=22 THEN F=0
440 IF F=0 THEN LOCATE 1,1:PRINT "SCORE
";SC:GOTO 500
450 LOCATE BX,BY:PRINT " ";BY=BY+1:IF B
Y>=23 THEN F=0:GOTO 500
460 TE=TEST((16*BX)-2,(16*(25-BY))+3)
470 IF TE=1 THEN SC=SC+1
480 LOCATE BX,BY:PRINT CHR$(252)
490 SOUND 1,100+(BY*30),7,15,1
500 NEXT X:LOCATE X,Y:PRINT " ";
510 NEXT Y
520 GOTO 920
530 REM *****
540 REM * EXPLOSION *
550 REM *****
560 X=X+1
570 LOCATE X,Y:PRINT " "
580 FOR I=Y TO 23
590 FOR J=15 TO 0 STEP -2
600 TAG
610 PLOT (x-2)*16,(27-i)*14+j,3
620 SOUND 1,500-((23-i)*16+j),4,15
630 PRINT "[";
640 INK 3,INT(RND(TIME)*29)+1
650 NEXT j
660 NEXT i
670 SOUND 1,350,1000,1,1,1,21
680 FOR i=1 TO 100
690 INK 3,RND(TIME)*29
700 PLOT (x-2)*16,46,3:DRAW (x-2)*16+(IN
T(RND(TIME)*100-50)),46+RND(TIME)*90,3
710 NEXT i
720 GOTO 1230
730 REM *****
740 REM * Draw new city *
750 REM *****
760 PEN 1

```

```

770 FOR i=7 TO 33
780 X=INT(RND(TIME)*LV)+1
790 FOR j=1 TO X
800 LOCATE i,23-j
810 PRINT "j";
820 NEXT j
830 LOCATE i,23-j
840 PRINT "\"
850 NEXT i
860 PEN 2
870 LOCATE 1,23:PRINT STRING$(40,143)
880 RETURN
890 REM *****
900 REM * LANDED!! GO FOR A WALK *
910 REM *****
920 LOCATE X,Y-1:PRINT "[ "
930 FOR I=(X+20)*10-20 TO X*10 STEP -5
940 TAB
950 PLOT I,63,3:PRINT CHR$(250+F);" ";:F
=1-F
960 FOR J=1 TO 60:NEXT J:SOUND 1,100,1,1
5
970 NEXT I
980 REM *****
990 REM * YAHOOO! *
1000 REM *****
1010 FOR K=1 TO 30
1020 PLOT I,63:PRINT CHR$(248+F);" ";:F=
1-F
1030 FOR J=1 TO 60:NEXT J
1040 NEXT K
1050 REM *****
1060 REM * NEXT LEVEL *
1070 REM *****
1080 MODE 0
1090 PEN 2
1100 LV=LV+3:IF LV>17 THEN LV=17
1110 LOCATE 6,7:PRINT"NEXT LEVEL"


```

```

1120 FOR I=100 TO 0 STEP -4
1130 FOR J=I TO 0 STEP -7
1140 SOUND 1,J+10,2,15
1150 INK 2,J/4
1160 NEXT J
1170 NEXT I
1180 INK 2,21
1190 GOTO 340
1200 REM *****
1210 REM * GAME OVER *
1220 REM *****
1230 SOUND 131,100,10,10:MODE 0:FOR I=1
TO 15:INK I,I:NEXT I:INK 0,0
1240 FOR I=1 TO 15
1250 SOUND 1,10*I,3,15
1260 LOCATE 7,I:PEN I:PRINT"GAME OVER"
1270 FOR J=1 TO 30:NEXT J
1280 NEXT I
1290 FOR I=1 TO 14
1300 LOCATE 7,I:PEN 0:PRINT " "
1310 FOR J=1 TO 30:NEXT J
1320 NEXT I:PEN 15
1330 PRINT:PRINT" FINAL SCORE ";SC
1340 PEN 12
1350 PRINT " PLAY AGAIN ? (Y/N)"
1360 a$=INKEY$
1370 IF a$="y" THEN RUN
1380 IF a$<>"n" THEN 1370
1390 MODE 1:PEN2:END

```


Strafe



You are coming in for a low-level bombing run and your orders are to knock out everything in sight. Don't forget that you are running out of fuel as well, so look for fuel dumps. The space bar is your bombing release mechanism.

```
100 REM *****
110 REM * Strafe *
120 REM *****
130 MODE 1
140 ENT 1,100,-2,2
150 ENV 1,100,2,2
160 INK 1,22:INK 0,0:PAPER 0:PEN 1
170 INK 2,18:INK 3,6
180 REM *****
190 REM * Instructions *
200 REM *****
```

```
210 LOCATE 17,1:PRINT "Strafe"
220 PRINT
230 PRINT "You have just flown past enem
y lines."
240 PRINT "and have received orders to b
omb every"
250 PRINT "thing in sight."
260 PRINT "The targets will consist of..
."
270 PRINT
280 PEN 2
290 PRINT
300 PRINT "Houses          These will score
10 points"
310 PRINT
320 PRINT "Radar          These will score
20 points"
330 PRINT
340 PRINT "Fuel dumps     These will give y
ou an"
350 PRINT "                20 units of fuel"
360 PEN 1
370 PRINT
380 PRINT "Use the SPACE BAR to drop you
r bombs"
390 PEN 3
400 PRINT
410 PRINT "      Press any key to start bo
mbing"
420 IF INKEY$="" THEN 420
430 y=3
440 e=-1
450 REM *****
460 REM * U.D.G.'S *
470 REM *****
480 SYMBOL AFTER 91
490 SYMBOL 91,24,70,223,207,94,24,90,165
```

```

500 SYMBOL 92,24,60,126,255,211,255,219,
251
510 SYMBOL 93,4,124,72,76,126,223,31,63
520 SYMBOL 94,0,128,192,224,255,248,127,
0
530 SYMBOL 95,0,0,96,144,254,255,254,0
540 SYMBOL 231,0,0,24,60,60,24,0,0
550 SYMBOL 248,0,0,0,0,0,24,28,62
560 SYMBOL 247,0,0,0,24,28,62,127,255
570 SYMBOL 246,60,126,127,255,126,60,24,
24
580 SYMBOL 245,0,60,126,255,60,24,24,24
590 SYMBOL 244,0,24,60,127,28,24,8,24
600 SYMBOL 243,0,0,24,60,127,28,24,28
610 SYMBOL 242,0,0,0,8,24,56,28,16
620 SYMBOL 241,0,0,0,0,8,24,56,28
630 SYMBOL 240,0,0,0,0,0,0,0,24
640 FOR i=0 TO 3:INK i,0:NEXT i
650 MODE 1
660 PRINT CHR$(22);"1"
670 GOSUB 1460
680 PRINT CHR$(22);"0"
690 GOSUB 1760
700 c=3
710 FOR fu=0 TO 201 STEP 2:GOSUB 1710:NE
XT fu
720 sh1$=" "
730 sh2$=CHR$(94)+CHR$(95)
740 PAPER 2
750 PEN 3:GOSUB 1020:PEN 1
760 PEN 2:LOCATE 1,14:PRINT STRING$(40,1
43)
770 PEN 1
780 c$="2"
790 FOR i=1 TO 40
800 GOSUB 1290
810 l$=l$+c$
820 NEXT i

```

```

830 INK 0,11:INK 2,18:INK 3,2:INK 1,0
840 PAPER 2
850 c=0
860 REM *****
870 REM * start *
880 REM *****
890 LOCATE 1,13:PRINT 1$
900 GOSUB 1290
910 1$=RIGHT$(1$,39)+c$
920 IF INKEY$=" " AND f=0 AND fu>0 THEN
f=1:fx=21:fy=y+2
930 IF f=1 THEN GOSUB 1090
940 IF e<>-1 THEN GOSUB 1380
950 IF fu=0 THEN y=y+1:IF y>12 THEN 1900
960 IF fu=0 THEN GOSUB 1020
970 IF fu>0 THEN GOSUB 1700
980 GOTO 890
990 REM *****
1000 REM * Draw plane *
1010 REM *****
1020 LOCATE 19,y:PRINT sh1$;
1030 LOCATE 19,y+1:PRINT sh2$;
1040 TAGOFF
1050 RETURN
1060 REM *****
1070 REM * Drop bomb *
1080 REM *****
1090 LOCATE fx,fy:PRINT " "
1100 fx=fx-1:fy=fy+1
1110 s$=MID$(1$,fx,1)
1120 IF fy<>14 THEN 1230
1130 IF s$="[" THEN 1180
1140 IF s$="]" THEN sc=sc+20:GOSUB 1830
1150 IF s$="\ " THEN sc=sc+10:GOSUB 1830
1160 IF s$<>" " THEN e=8:pt=fx:f=0:RETURN
1170 GOTO 1230
1180 fu=fu+20:IF fu>198 THEN fu=198
1190 SOUND 1,184,200,1,1,1

```

```


1200 c=3:i=fu :FOR fu=1 TO i STEP 2:GOSU
B 1710:NEXT fu
1210 fu=i
1220 c=0:GOTO 1160
1230 IF fy=14 THEN f=0:RETURN
1240 LOCATE fx,fy:PRINT CHR$(231)
1250 RETURN
1260 REM *****
1270 REM * Next bit of landscape *
1280 REM *****
1290 a=INT (20*RND(TIME)+1)
1300 IF a<14 THEN c$=" "
1310 IF a<16 AND a>14 THEN c$="]"
1320 IF a<19 AND a>16 THEN c$="\ "
1330 IF a=20 THEN c$="["
1340 RETURN
1350 REM *****
1360 REM * Explosion *
1370 REM *****
1380 pt=pt-1
1390 l$=LEFT$(l$,pt)+CHR$(240+e)+RIGHT$(
l$, (39-pt))
1400 e=e-1:IF e=-1 THEN RETURN
1410 SOUND 1,550+pt*10+e,8,15,2,2,e+3
1420 RETURN
1430 REM *****
1440 REM * Draw mountains *
1450 REM *****
1460 PEN 2
1470 FOR i=1 TO 13
1480 LOCATE 1,i:PRINT STRING$(i,143);CHR
$(215)
1490 LOCATE 16,i:PRINT STRING$(i*2,143);
CHR$(215)
1500 LOCATE 40-i,i
1510 PRINT CHR$(214);
1520 PRINT STRING$(i,143)
1530 LOCATE 16-i,i:PRINT CHR$(214);STRIN
G$(i,143)

```

```
1540 NEXT i
1550 PAPER 0: PEN 1
1560 PLOT 500,280,1: DRAW 420,200
1570 PLOT 128,290: DRAW 32,196
1580 i1=300
1590 x1=400
1600 FOR i=300 TO 200 STEP -15
1610 xx=INT (40*RND(TIME)+390)
1620 aa=aa+2
1630 FOR j=1 TO aa: PLOT xx+j*2,i,0: DRAW
x1+j*2,i1: NEXT j
1640 x1=xx: i1=i
1650 NEXT i
1660 RETURN
1670 REM *****
1680 REM * Fuel *
1690 REM *****
1700 fu=fu-1
1710 PLOT 300+fu,120,c: DRAW 300+fu,150
1720 RETURN
1730 REM *****
1740 REM * Fuel box *
1750 REM *****
1760 PLOT 296,116,1: DRAW 296,154: DRAW 50
4,154
1770 DRAW 504,116: DRAW 296,116
1780 LOCATE 24,20: PRINT "FUEL"
1790 RETURN
1800 REM *****
1810 REM * Display score *
1820 REM *****
1830 PAPER 0
1840 LOCATE 2,18: PRINT "SCORE ="; sc
1850 PAPER 2
1860 RETURN
1870 REM *****
1880 REM * Game over *
1890 REM *****
1900 GOSUB 1020
```

```
1910 FOR i=1 TO 100
1920 PLOT 16*19,176,3
1930 SOUND 1,100+RND(TIME)*50,1,15,2,2,R
ND(TIME)*14+1
1940 DRAW 16*19+INT(150*RND(TIME)-75),17
6+INT(59*RND(TIME)+50)
1950 INK 3,RND (TIME)*25
1960 NEXT i
1970 PAPER 0:INK 0,0:INK 1,15
1980 INK 2,2
1990 INK 3,13
2000 CLS
2010 LOCATE 15,2:PRINT "Out of fuel"
2020 PEN 2
2030 LOCATE 16,6:PRINT "Game over "
2040 PEN 3
2050 LOCATE 12,11:PRINT "Final score =";
SC
2060 LOCATE 16,14:PRINT "Play again?"
2070 a$=INKEY$
2080 IF a$="" THEN 2070
2090 IF a$="y" THEN RUN
2100 IF a$="n" THEN END
2110 GOTO 2070
```

Mortar



Try to lob a mortar shell at the on-coming tank. Use the ↑ and ↓ keys to control the angle of the mortar, and the → to fire. Just one small problem though – knock out any protected buildings and you are in trouble.

```
100 REM *****
110 REM * Mortar *
120 REM *****
130 MODE 1
140 DEG
150 INK 0,0
160 INK 1,24,0
170 INK 2,22
180 INK 3,0,24
190 LOCATE 17,1
200 SPEED INK 10,10
210 PEN 1
```



```
220 REM *****
230 REM * Instructions *
240 REM *****
250 PRINT "Mortar"
260 PRINT
270 PEN 2
280 PRINT "An Enemy tank has broken thro
ugh your"
290 PRINT "defences,And is heading for y
our Capital"
300 PRINT "as a last resort a mortar has
  been set"
310 PRINT "up for you to use to try and
stop the"
320 PRINT "tank."
330 PRINT "When a shell is loaded into t
he mortar"
340 PRINT "the power will be displayed a
t the "
350 PRINT "bottom of the screen."
360 PRINT "Next to the power will be a d
isplay"
370 PRINT "for the angle."
380 PRINT "Using the up and down cursor
controls"
390 PRINT "adjust the angle of the morta
r."
400 PRINT "When you think the trajectory
of the"
410 PRINT "bomb is correct then press th
e cursor"
420 PRINT "space bar."
430 PEN 3
440 PRINT
450 PRINT "      Press any key to start"
460 IF INKEY$="" THEN GOTO 460
470 REM *****
480 REM * Initialize *
490 REM *****
```

```
500 MODE 0
510 SYMBOL AFTER 91
520 SYMBOL 91,0,62,254,28,126,129,129,12
6
530 angle=45
540 tank =640
550 TAG:PLOT tank,49,9:PRINT CHR$(91);"
";:TAGOFF
560 g=24
570 GOSUB 1430
580 power =INT(50*RND(TIME)+90)
590 GOTO 700
600 PEN 6
610 PAPER 5
620 LOCATE 1,24:PRINT "Power = ";power
630 PEN 7
640 LOCATE 1,25:PRINT "Angle = ";angle
650 FOR i=1 TO 30:a$=INKEY$:NEXT i
660 PAPER 0
670 a$=INKEY$
680 IF a$="" THEN GOTO 670
690 IF a$=" " THEN 810
700 PLOT 0,35,0
710 DRAW SIN(90-angle)*20,COS(90-angle)*
20+35
720 PLOT 0,35,0
730 IF a$=CHR$(241) AND angle >0 THEN an
gle=angle-1
740 IF a$=CHR$(240) AND angle <90 THEN a
ngle=angle+1
750 PLOT 0,35,8
760 DRAW SIN(90-angle)*20,COS(90-angle)*
20+35
770 GOTO 600
780 REM *****
790 REM * Main loop *
800 REM *****
810 DEG
820 snd=550
```

```

830 r=power^2*SIN(2/angle)/g
840 h=(power*SIN(angle))^2/(2*g)
850 st=2*power*SIN(angle)/g
860 IF st=0 THEN st=1
870 FOR t=0 TO st STEP st/100
880 x1=x
890 y1=y
900 x=INT(power*COS(angle)*t)
910 y1=y
920 y=INT((power*SIN(angle)-g*t/2)*t)
930 snd=snd+(y1-y)
940 SOUND 1,snd,10,15
950 IF TEST(x,y+35)=1 THEN GOTO 1330
960 IF TEST(x,y+35)=9 THEN GOTO 2030
970 IF TEST(x,y+35)=10 THEN GOTO 2060
980 PLOT x,35+y,5
990 PLOT x1,35+y1,0
1000 NEXT t
1010 REM *****
1020 REM * Explosion *
1030 REM *****
1040 c=0
1050 LOCATE 1,1
1060 PRINT CHR$(23);"1"
1070 FOR i=x-25 TO x+25
1080 PLOT x,35,c:DRAW i,80
1090 c=c+0.3
1100 SOUND 1,500+i,2,15,1,1,8
1110 NEXT i
1120 c=0
1130 FOR i=x-25 TO x+25
1140 PLOT x,35,c:DRAW i,80
1150 c=c+0.3
1160 NEXT i
1170 LOCATE 1,1
1180 PRINT CHR$(23);"0"
1190 REM *****
1200 REM * Move tank *
1210 REM *****

```

```

1220 FOR i=1 TO 20
1230 tank=tank-1
1240 TAG:PLOT tank,49,9:PRINT CHR$(91);"
";:TAGOFF
1250 IF TEST(tank-10,37)=10 THEN xx=v1+3
0:GOSUB 1860:GOTO 2100
1260 NEXT i
1270 PLOT x,y+35,0:PLOT x2,y1+35,0
1280 GOTO 580
1290 END
1300 REM *****
1310 REM * Blow hole in mountain *
1320 REM *****
1330 PLOT x,y+35,0:PLOT x1,y1+35,0
1340 ORIGIN x-10,35+y-10
1350 FOR i=1 TO 30
1360 PLOT INT(20*RND(TIME)+1),INT(20*RND
(TIME)+1),0
1370 NEXT i
1380 ORIGIN 0,0
1390 GOTO 1220
1400 REM *****
1410 REM * Draw up screen display *
1420 REM *****
1430 INK 0,11
1440 INK 1,18:REM Hills and grass
1450 INK 2,0
1460 INK 5,0:REM Bomb
1470 INK 6,2
1480 INK 7,6
1490 INK 8,26
1500 INK 9,9
1510 INK 10,3
1520 INK 11,6,15
1530 INK 12,24,6
1540 INK 13,15,24
1550 SPEED INK 1,1
1560 x=150
1570 y=INT(60*RND(TIME)+50)

```


```
1580 PLOT x,33,1:DRAW x,33+y
1590 PLOT 150+(150-x),33:DRAW 150+(150-x
),33+y
1600 y=y-RND(TIME)
1610 IF y<0 THEN GOTO 1630
1620 x=x-0.5:GOTO 1580
1630 PLOT 0,33:DRAW 640,33
1640 PAPER 5
1650 FOR i=24 TO 25
1660 LOCATE 1,i:PRINT SPACE$(20);
1670 NEXT i
1680 PAPER 0
1690 v1=150+(150-x)+20
1700 ORIGIN v1,35
1710 FOR i=1 TO 20
1720 PLOT i,0,10:DRAW i,15
1730 PLOT 20+i/2,0:DRAW 20+i/2,20:DRAW 2
5,37
1740 NEXT i
1750 FOR i=1 TO 36 STEP 6
1760 h=INT(5*RND(TIME)+2):h=h*3
1770 FOR j=1 TO 6
1780 PLOT 34+i*2+j,0:DRAW 34+i*2+j,h
1790 NEXT j
1800 NEXT i
1810 ORIGIN 0,0
1820 RETURN
1830 REM *****
1840 REM * Nuke *
1850 REM *****
1860 c=11
1870 RAD
1880 FOR i=0 TO 40
1890 PLOT xx,i+35,c:DRAW xx+35,i+35
1900 c=c+1:IF c=14 THEN c=11
1910 NEXT i
1920 FOR i=0 TO PI*2 STEP 0.1
1930 x=cos(i)*40:y=sin(i)*30
```

```
1940 PLOT xx+x+20,90+y,c
1950 DRAW xx-x+20,90+y
1960 SOUND 1,200+i*x,2,15,1,1,15
1970 c=c+1:IF c=14 THEN c=11
1980 NEXT i
1990 FOR i=1 TO 1500:NEXT i
2000 MODE 1:PAPER 0:INK 0,0:PEN 1:INK 1,
6
2010 INK 2,20
2020 RETURN
2030 xx=tank+4:GOSUB 1860:LOCATE 13,1
2040 PRINT "You hit the tank"
2050 GOTO 2120
2060 xx=v1+15:GOSUB 1860
2070 LOCATE 10,1
2080 PRINT "You blew the village up"
2090 GOTO 2120
2100 LOCATE 10,1
2110 PRINT "The tank hit the city"
2120 LOCATE 16,10:PEN 2
2130 PRINT "Game over"
2140 LOCATE 11,12:PEN 3:INK 3,24
2150 PRINT "Final score =";tank-v1
2160 LOCATE 13,14:PEN 1
2170 PRINT "Play again ?";
2180 a$=INKEY$
2190 IF a$="y" THEN PRINT "Yes":FOR i=1
TO 1000:NEXT i:RUN
2200 IF a$="n" THEN PRINT "No":FOR i=1 T
O 1000:NEXT i:END
2210 GOTO 2180
```


Chapter 6

MODERN GAMES

Batter



You have seven balls to try and batter your way through my wall. Use the ← and → keys to control the bat. If you get through, you get an extra ball each time. Try to build a big score.

```
100 REM *****
110 REM * Batter *
120 REM *****
130 MODE 1
140 INK 1,6:INK 2,2:INK 3,24
150 DIM a(5,19)
160 PAPER 0:INK 0,0
170 REM *****
180 REM * Instructions *
190 REM *****
200 LOCATE 17,1
210 PEN 1
```

```
220 PRINT "Batter"
230 PRINT CHR$(22);"1"
240 LOCATE 17,1
250 PEN 2
260 PRINT "_____"
270 PRINT CHR$(22);"0"
280 PRINT
290 PRINT "Controls"
300 PRINT
310 PRINT "<Cursor left> Move bat to th
e left"
320 PRINT "<Cursor right> Move bat to th
e right"
330 PRINT "Knock the bricks out of the w
all by"
340 PRINT "directing the ball with the b
at."
350 PRINT
360 PRINT "<1> batter"
370 PRINT "<2> batthrough"
380 PRINT "<3> 'who cares ???'"
390 PEN 3
400 PRINT
410 PRINT " Press 1,2 or 3"
420 a$=INKEY$
430 IF a$="" THEN 420
440 flag=1
450 f=0
460 IF a$="1" THEN flag=1:GOTO 500
470 IF a$="2" THEN flag=0:GOTO 500
480 IF a$="3" THEN f=1:GOTO 500
490 GOTO 420
500 MODE 0
510 sc=0:bl=7
520 INK 1,2:INK 2,6:INK 3,24:INK 4,18
530 INK 5,20:INK 6,6:INK 7,8:INK 8,15
540 INK 9,9
550 INK 10,26,2
560 FOR i=3 TO 7
```

```
570 FOR j=2 TO 19
580 a(i-2,j)=1
590 NEXT j:NEXT i
600 SPEED KEY 1,1
610 GOSUB 1060
620 cn=0
630 x=15
640 bx=INT(14*RND(TIME)+3)
650 by=10
660 bxd=-1
670 byd=-1
680 REM *****
690 REM * Main loop *
700 REM *****
710 GOSUB 770
720 GOSUB 870
730 GOTO 710
740 REM *****
750 REM * Move Bat *
760 REM *****
770 a$=INKEY$
780 IF a$=CHR$(242) AND x>1 THEN x=x-1
790 IF a$=CHR$(243) AND x<16 THEN x=x+1
800 LOCATE x,25
810 PEN 9
820 PRINT " ";STRING$(3,206);" ";
830 RETURN
840 REM *****
850 REM * Move ball *
860 REM *****
870 x1=bx
880 y1=by
890 bx=bx+bxd
900 IF bx<3 OR bx>18 THEN bxd=-bxd
910 by=by+byd
920 IF by<3 THEN byd=-byd
930 IF by>23 AND (bx<x OR bx>x+4) THEN G
OTO 1480
```


```
940 IF by>23 THEN byd=-byd:IF bx>3 AND b
x<17 THEN bx=bx+bx d
950 IF by>2 AND by<8 THEN xx=8-by:GOSUB
1280
960 LOCATE x1,y1
970 PRINT " "
980 LOCATE bx,by
990 PEN 8
1000 INK 8,INT(24*RND(TIME)+1)
1010 PRINT CHR$(230)
1020 RETURN
1030 REM *****
1040 REM * Draw wall *
1050 REM *****
1060 FOR i=3 TO 7
1070 FOR j=19 TO 2 STEP -1
1080 IF a(i-2,j)=0 THEN GOTO 1130
1090 SOUND 1,j*(i/2),1,10
1100 PEN i
1110 a=10-i:LOCATE j,a
1120 PRINT CHR$(143)
1130 NEXT j
1140 NEXT i
1150 PEN 1
1160 FOR i=24 TO 1 STEP -1
1170 LOCATE 1,i
1180 PRINT CHR$(143)
1190 LOCATE 20,i
1200 PRINT CHR$(143)
1210 SOUND 1,i*10,2,10
1220 NEXT i
1230 LOCATE 1,1:PRINT STRING$(20,143);
1240 RETURN
1250 REM *****
1260 REM * Update wall status *
1270 REM *****
1280 yy=INT(bx)
1290 IF a(xx,yy)=0 THEN 1380
```

```
1300 IF f=1 THEN flag =INT(RND(TIME)+1)
1310 IF flag=1 THEN byd=-byd:GOTO 1320
1320 sc=sc+xx
1330 a(xx,yy)=0
1340 SOUND 1,xx*100,4,10
1350 cn=cn+1
1360 IF cn=90 THEN GOSUB 1640:b1=b+1:GOT
O 560
1370 GOSUB 1420
1380 RETURN
1390 REM *****
1400 REM * Score *
1410 REM *****
1420 PAPER 1
1430 PEN 10
1440 LOCATE 1,1:PRINT "score =";sc
1450 LOCATE 14,1:PRINT "ball";b1
1460 PAPER 0
1470 RETURN
1480 b1=b1-1
1490 GOSUB 1640
1500 IF b1=0 THEN 1530
1510 GOSUB 1420
1520 GOTO 630
1530 PEN 2
1540 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
1550 PRINT "You are out of balls"
1560 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
:PRINT:PRINT:PRINT:PRINT
1570 PEN 4
1580 PRINT "      Play again ?"
1590 PRINT:PRINT:PRINT:PRINT
1600 a$=INKEY$
1610 IF a$="y" THEN RUN
1620 IF a$="n" THEN END
1630 GOTO 1600
1640 FOR i=1 TO 20 :SOUND 1,i*5+(i/2),8,
10:NEXT i
```

```
1650 CLS:GOSUB 1060
```

```
1660 RETURN
```

Colours



This one will really try you out. You have to get through a set of coloured bricks using ←, → and ↑. You can only go up if the path is the same colour as the brick at the top. It all gets faster with difficulty level but as well as the colour of the top bricks changing, now they are all different colours.

```
100 REM *****
110 REM * Colours *
120 REM *****
130 REM
140 MODE 0 : PAPER 0 : INK 0,0
150 CLS : INK 1,24
160 LOCATE 1,1: PEN 2
170 PRINT "      Colours"
180 PRINT "      -----"
190 PRINT: PEN 1
```



```
200 PRINT " Use the arrow key"
210 PRINT "to get to the top"
220 PRINT "You can only move"
230 PRINT "up if the square"
240 PRINT "above you is the "
250 PRINT "same colour as the"
260 PRINT "one at the top."
270 PEN 2:INK 2,5
280 PRINT:PRINT
290 PRINT"Press a key to play"
300 IF INKEY$=""THEN GOTO 300
310 DIM maze(10,6),top(6)
320 MODE 0:INK 0,0
330 PAPER 0:BORDER 0
340 INK 1, 2:INK 2, 6:INK 3,18
350 INK 4,26:INK 5,24:INK 6,17
360 clock = 300
370 PEN 1
380 REM
390 REM *****
400 REM * Set Up Board *
410 REM *****
420 REM
430 px = 10 : py = 10
440 FOR i = 1 TO 10
450   FOR j = 1 TO 6
460     maze(i,j)=0
470   NEXT j
480 NEXT i
490 FOR i = 1 TO 10
500   FOR j=1 TO 6
510     place=INT(6*RND(1))+1
520     IF maze(i,place)<>0 THEN GOTO 510
530     maze (i,place)=j
540   NEXT j
550 NEXT i
560 REM
```

```

570 REM *****
580 REM * Print Board *
590 REM *****
600 REM
610 DI
620 FOR i = 1 TO 10
630 FOR j = 1 TO 6
640 LOCATE (j-1)*3+2,i+5
650 PEN maze(i,j)
660 PRINT STRING$(3,143);
670 NEXT j
680 NEXT i
690 LOCATE px,py+5
700 PAPER maze(py,INT((px+1)/3))
710 PEN 0 : PRINT CHR$(255);
720 EI
730 GOSUB 1330
740 EVERY 300-level*5 GOSUB 1330
750 REM
760 REM *****
770 REM * Main Loop *
780 REM *****
790 REM
800 DI:a$=INKEY$
810 clock = clock -1
820 LOCATE 5,18 : PAPER 4 : PEN 0
830 PRINT "Time = ";clock
840 IF clock = 0 THEN GOTO 1520
850 EI
860 IF a$="" THEN GOTO 800
870 IF a$<CHR$(240) OR a$>CHR$(243) THEN
GOTO 800
880 DI
890 above=maze(py-1,INT((px+1)/3))
900 under=maze(py,INT((px+1)/3))
910 IF a$<>CHR$(242) THEN GOTO 1020
920 IF px < 3 THEN GOTO 800
930 LOCATE px,py+5

```


```

940     PAPER under:PEN under
950     PRINT CHR$(143)
960     px = px -1
970     under=maze(py,INT((px+1)/3))
980     LOCATE px,py+5
990     PAPER under:PEN 0
1000    PRINT CHR$(255);
1010    GOTO 800
1020 IF a$<>CHR$(243) THEN GOTO 1130
1030   IF px > 18 THEN GOTO 800
1040     LOCATE px,py+5
1050     PEN under
1060     PRINT CHR$(143);
1070     px = px +1
1080     under=maze(py,INT((px+1)/3))
1090     LOCATE px,py+5
1100     PAPER under:PEN 0
1110     PRINT CHR$(255);
1120     GOTO 800
1130 IF above<>top(INT((px+1)/3)) THEN
GOTO 800
1140   LOCATE px,py+5
1150   PEN under
1160   PRINT CHR$(143);
1170   py = py -1
1180   under=maze(py,INT((px+1)/3))
1190   LOCATE px,py+5
1200   PAPER under:PEN 0
1210   PRINT CHR$(255);
1220   IF py>1 THEN GOTO 800
1230 level = level + 1
1240 EVERY 5000 GOSUB 1330
1250 EI
1260 GOTO 320
1270 REM
1280 REM
1290 REM *****
1300 REM * Draw Top Line *
1310 REM *****

```

```
1320 REM
1330 DI:IF level > 5 THEN GOTO 1420
1340 place=INT(6*RND(1))+1
1350 FOR i = 1 TO 6
1360 top(i)=place
1370 NEXT i
1380 LOCATE 2,1
1390 PEN place
1400 PRINT STRING$(18,143);
1410 EI:RETURN
1420 FOR i = 1 TO 6:top(i)=0:NEXT i
1430 FOR i = 1 TO 6
1440 place=INT(6*RND(1))+1
1450 IF top(place)<>0 THEN GOTO 1440
1460 top(place)=i
1470 LOCATE (place-1)*3+2,1
1480 PEN top(place)
1490 PRINT STRING$(3,143);
1500 NEXT i
1510 EI:RETURN
1520 LOCATE 6,1
1530 PEN 0:PAPER 4
1540 PRINT "Out of time"
1550 END
```

Demon Driver



Have you got what it takes to become a top racing driver? Zip round the course as fast as you can and as the difficulty level goes up so does your speed - but the road gets narrower. The controls are ← and →

```
100 REM *****
110 REM * Demon driver *
120 REM *****
130 MODE 1
140 INK 1,6:INK 2,17:INK 3,18
150 PEN 1
160 LOCATE 14,1:PRINT "Demon driver"
170 PRINT
180 PRINT "Drive your racing car down the road"
190 PRINT "for as long as possible."
```

```


200 PRINT "If you hit the sides of the r
oad you "
210 PRINT "will crash and a rating and s
core will "
220 PRINT "given."
230 PRINT
240 PEN 2
250 PRINT "controls "
260 PRINT "Use left and right cursor con
trols"
270 PRINT "for steering."
280 PRINT
290 PRINT
300 PEN 3
310 PRINT "press any key to start drive.
"
320 IF INKEY$="" THEN GOTO 320
330 SYMBOL AFTER 91
340 SYMBOL 91,90,126,90,24,24,219,255,21
9
350 t=15
360 ct=22
370 d=100
380 wd=12
390 m=1
400 MODE 1
410 INK 0,0
420 INK 1,13
430 INK 2,18
440 INK 3,6
450 GOSUB 480
460 FOR i=1 TO 25:LOCATE t,i:PRINT t$:NE
XT i
470 GOTO 550
480 t$=CHR$(206)+CHR$(206)
490 t$=t$+CHR$(14)+"1"
500 wd=wd-1:IF wd=3 THEN wd=4
510 t$=t$+SPACE$(wd)

```

```
520 t$=t$+CHR$(14)+"2"
530 t$=t$+CHR$(206)+CHR$(206)
540 RETURN
550 GOSUB 480
560 REM
570 PEN 3
580 PAPER 2
590 LOCATE t,1
600 PRINT CHR$(11);
610 PRINT t$
620 PEN 0:PAPER 1:LOCATE ct,25:PRINT CHR
$(91)
630 t=t+INT(3*RND(TIME)-1)
640 IF t=0 THEN t=1
650 IF t>22 THEN t=22
660 d=d-1:IF d=0 THEN d=100:GOSUB 480
670 SOUND 1,2000,6,10,2,2,2
680 m=m+1
690 a$=INKEY$
700 IF a$=CHR$(242) AND ct>0 THEN ct=ct-
1
710 IF a$=CHR$(243) AND ct<40 THEN ct=ct
+1
720 ctx=ct*16-8:cty=20
730 IF TEST(ctx,cty)=1 THEN GOTO 560
740 LOCATE ct+1,25:PRINT " ":LOCATE ct-1
,25:PRINT " "
750 FOR i=1 TO 100
760 LOCATE ct,25:PRINT CHR$(238)
770 SOUND 1,i*10+2000,1,10,1,1,5
780 LOCATE ct,25:PRINT CHR$(230)
790 INK 0,RND(TIME)*24
800 NEXT i
810 REM *****
820 REM * Game over *
830 REM *****
840 INK 0,0:PAPER 0
850 INK 1,6
```

```
860 INK 2,6
870 INK 3,24
880 PEN 3
890 CLS
900 LOCATE 16,3:PRINT "Game over"
910 PRINT
920 PEN 1
930 PRINT "Rating is :-";
940 IF m>3000 THEN PEN 3:PRINT "Demon dr
iver":GOTO 1020
950 IF m>2000 THEN PRINT "Pro":GOTO 1020
960 IF m>1000 THEN PRINT "Amatuer":GOTO
1020
970 IF m>500 THEN PRINT "So So":GOTO 102
0
980 IF m>250 THEN PRINT "Rubbish":GOTO 1
020
990 IF m>100 THEN PRINT "Woman driver":G
OTO 1020
1000 IF m>25 THEN PRINT "You need glasse
s":GOTO 1020
1010 IF m<25 THEN PRINT "A 2 year old co
uld do better"
1020 LOCATE 13,7:PRINT "Final Milage =";
m
1030 PEN 3
1040 LOCATE 15,10:PRINT "Play again ?"
1050 a$=INKEY$
1060 IF a$="y" THEN RUN
1070 IF a$="n" THEN END
1080 GOTO 1050
```


Docking



Avoid all the asteroids and space garbage. Dock successfully with the mother ship before your fuel runs out. Cursor keys control your thrusters.

```
100 REM *****
110 REM * Docking *
120 REM *****
130 MODE 1
140 LOCATE 17,1:PRINT "Docking"
150 PRINT
160 PRINT "Try to dock you shuttle craft
with the"
170 PRINT "Mother ship at the top of the
screen"
180 PRINT "The only snag is that there a
re 3 "
190 PRINT "Asteroid belts to avoid."
```

```
200 PEN 2
210 PRINT "the only controls for your shuttle"
220 PRINT "craft are the 3 cursor control keys"
230 PRINT "Right ,Left and Up."
240 PRINT "if you hit an asteroid or run out"
250 PRINT "fuel the game will end and a score given"
260 PRINT
270 PEN 3
280 PRINT "  Press any key to start."
290 IF INKEY$="" THEN 290
300 REM *****
310 REM * Initialize *
320 REM *****
330 INK 0,0:PAPER 0:BORDER 0
340 SYMBOL AFTER 91
350 INK 2,6:INK 1,13:INK 3,1
360 fum=300
370 RAD
380 SYMBOL 93,56,124,119,255,187,254,60,
28
390 SYMBOL 92,24,24,60,60,126,126,255,25
5
400 PEN 3
410 SYMBOL 91,60,126,255,255,126,36,66,1
29
420 MODE 1
430 x=20:y=25
440 pt=1
450 REM *****
460 REM * Draw space ship *
470 REM *****
480 FOR i=1 TO 100
490 PLOT i,400-i,1:DRAW 640-i,400-i
500 NEXT i
```




```
850 GOTO 590
860 REM *****
870 REM * Explosion *
880 REM *****
890 ty=408-(y*16)
900 tx=x*16-8
910 FOR i=1 TO 360 STEP 2
920 cc=i/5
930 PLOT tx+SIN(i)*cc,ty+COS(i)*cc,RND(T
IME)*3
940 SOUND 1,i+1000,3,15,1,1,4
950 NEXT i
960 GOTO 1110
970 REM *****
980 REM * Docked *
990 REM *****
1000 MODE 1
1010 LOCATE 15,1:PRINT "You Docked"
1020 fum=fum-30
1030 IF fum<30 THEN fum=30
1040 LOCATE 3,5:PRINT "you now only have
";fum;"fuel units"
1050 FOR j=1 TO 1000:NEXT j:GOTO 420
1060 REM *****
1070 REM * Game over *
1080 REM *****
1090 MODE 1
1100 PEN 1
1110 LOCATE 16,2:PRINT "Game over"
1120 PEN 2
1130 LOCATE 13,6:PRINT "Final score=";sc
1140 PEN 3
1150 LOCATE 16,11:PRINT "Play again"
1160 a$=INKEY$
1170 IF a$="y" THEN RUN
1180 IF a$="n" THEN END
1190 GOTO 1160
```


Chapter 7

TRADITIONAL GAMES AND SPORTS

Horse Race



A chance to earn some money from your friends. Back your favourite depending on its form and odds. As with Hangman, you may want to add your own names. This is done by again altering the data statement.

```
100 REM *****
110 REM * Horse race *
120 REM *****
130 SYMBOL AFTER 91
140 SYMBOL 91,0,0,2,7,124,252,136,136
150 SYMBOL 92,0,0,0,2,127,252,136,80
160 MODE 0
170 PAPER 0: INK 0,0
180 INK 1,2
190 INK 2,6
200 INK 3,8
210 INK 4,18
```



```

220 INK 5,20
230 INK 6,24
240 INK 7,26
250 PEN 8:INK 8,6
260 INK 11,0
270 INK 14,22
280 LOCATE 1,1
290 REM *****
300 REM * Instructions *
310 REM *****
320 PRINT "    Horse race"
330 PRINT "    ====="
340 PEN 1
350 PRINT
360 PRINT "Bet on a horse using"
370 PRINT "the odds provided"
380 PRINT "Then enter amount to";
390 PRINT "bet."
400 PEN 9:INK 9,21
410 PRINT
420 PRINT "If a horse wins"
430 PRINT "odds go down."
440 PRINT
450 PRINT "If a horse looses"
460 PRINT "odds go up."
470 PEN 12:INK 12,2,20
480 PRINT "Press a key to start"
490 IF INKEY$="" THEN GOTO 490
500 REM *****
510 REM * Initialize *
520 REM *****
530 RESTORE 880
540 CLS
550 DIM a$(14),a(6),r(6)
560 DIM b(14),od(14)
570 money=200
580 FOR i=1 TO 14
590 READ a$(i)

```

```
600 od(i)=6
610 NEXT i
620 MODE 0:INK 0,0:PAPER 0
630 GOSUB 1240
640 FOR i=1 TO 6
650 a(i)=0
660 NEXT i
670 PEN 3
680 PRINT
690 PRINT "Which horse to bet on"
700 INPUT h
710 IF h<1 OR h>6 THEN h=INT(6*RND(TIME)
+1)
720 PRINT "you have chosen"
730 PEN h
740 PRINT a$(r(h))
750 PEN 14
760 PRINT "You have #";money
770 PRINT "How much to bet"
780 PRINT "1 to 25 pounds"
790 INPUT bet
800 IF bet <0 THEN bet =1
810 IF bet >25 THEN bet=25
820 IF money-bet<0 THEN GOTO 770
830 INK 0,9
840 CLS
850 REM *****
860 REM * Horses *
870 REM *****
880 DATA Discovery
890 DATA Tubular bells
900 DATA Flash AHH!!
910 DATA The N Team
920 DATA The Greaser
930 DATA Red Rum
940 DATA Vector Twin Turbo
950 DATA Ferrari
960 DATA Equinoxe
```


```
970 DATA Atarian
980 DATA Newtech
990 DATA Bone Shaker
1000 DATA The Defender
1010 DATA Amstrad
1020 REM *****
1030 REM * Race *
1040 REM *****
1050 PEN 0
1060 FOR i=1 TO 7
1070 PLOT 0,i*48-24,11:DRAW 640,i*48-24
1080 NEXT i
1090 PLOT 630,0:DRAW 630,310
1100 LOCATE 5,1:PEN 5:PRINT "Horse race"
1110 INK 0,9
1120 FOR i=1 TO 6
1130 x=INT(5*RND(TIME)+3)
1140 a(i)=a(i)+x
1150 IF a(i)>600 THEN 1450
1160 TAG :PLOT a(i),i*48,i:PRINT " ";CHR
$(91+s);
1170 s=1-s
1180 NEXT i
1190 s=1-s
1200 GOTO 1120
1210 REM *****
1220 REM * Display horses *
1230 REM *****
1240 FOR i=1 TO 14
1250 b(i)=0
1260 NEXT i
1270 FOR i=1 TO 6
1280 x=INT(14*RND(TIME)+1)
1290 IF b(x)=1 THEN GOTO 1280
1300 b(x)=1
1310 r(i)=x
1320 NEXT i
1330 c=0
```

```

1340 FOR i=1 TO 6
1350 c=c+1
1360 PEN c
1370 PRINT i;" ";a$(r(i))
1380 PRINT ,od(r(i));"- 1"
1390 PRINT
1400 NEXT i
1410 RETURN
1420 REM *****
1430 REM * Display who won *
1440 REM *****
1450 MODE 1:INK 0,14
1460 PEN 1
1470 LOCATE 18-LEN(a$(r(i)))/2,1:PRINT a
$(r(i));" Won"
1480 LOCATE 18-LEN(a$(r(h))),4
1490 PRINT "you bet on ";a$(r(h))
1500 IF h<>i THEN GOTO 1540
1510 LOCATE 15,6
1520 PEN 1:PRINT "You Won ";Od(r(i))*bet
1530 money=money+od(r(i))
1540 money=money-bet
1550 IF money<=0 THEN LOCATE 15,8:PRINT
"Out of money":END
1560 FOR j=1 TO 6:od(r(j))=od(r(j))+1
1570 NEXT j
1580 od(r(i))=od(r(i))-2
1590 IF od(r(i))<1 THEN od(r(i))=1
1600 REM *****
1610 REM * Play again *
1620 REM *****
1630 LOCATE 15,23
1640 PEN 2:PRINT "Race again ?"
1650 a$=INKEY$
1660 IF a$="y" THEN 620
1670 IF a$="n" THEN END
1680 GOTO 1650

```

Os and Xs



Moves are made using the cursor keys to select a box and then the copy key. You decide on who has the first move. If your move is illegal, the machine tells you. Try to beat the machine without a draw (stalemate).

```
100 REM *****
110 REM * O's and X's *
120 REM *****
130 REM
140 MODE 0 : INK 0,26 : CLS : BORDER 26
150 INK 1,26 : INK 2,26 : INK 3,26
160 INK 4,26 : INK 5,26 : INK 6,26
170 INK 7,26 : INK 8,26 : INK 9,26
180 INK 10,0 : INK 11,26 : INK 12,6
190 INK 13,21 : INK 14,20 : INK 15,0
200 PEN 15:CLS
```

```

210 INPUT "Do you want to go first ";g$
220 IF g$="" THEN GOTO 200
230 IF g$<>"n" AND g$<>"y" THEN 200
240 IF g$="y" THEN mo=1 ELSE mo=0
250 CLS
260 REM
270 REM *****
280 REM * Draw Board *
290 REM *****
300 REM
310 FOR i = 1 TO 2
320   FOR j = 3 TO 7 STEP 2
330     PLOT 150*i+j,0,15
340     DRAWR 0,400
350     PLOT 0,134*i+j
360     DRAWR 450,0
370   NEXT j
380 p=1
390 NEXT i
400 FOR i = 1 TO 3
410   FOR j = 1 TO 3
420     PLOT 146*i,130*j+j,p
430     DRAWR -125,0
440     DRAWR 0,-110
450     DRAWR 125,0
460     DRAWR 0,110
470     p=p+1
480   NEXT j
490 NEXT i
500 REM
510 REM *****
520 REM * Set Up Array For Pieces *
530 REM *****
540 REM
550 DIM array (3,3)
560 FOR i = 1 TO 3
570   FOR j = 1 TO 3
580     array(i,j) = 0

```

```

590 NEXT j
600 NEXT i
610 :
620 FOR goes=1 TO 9
630 IF mo=0 THEN GOTO 690
640 mo=0:GOSUB 800
650 array(x,y)=1
660 GOSUB 1090
670 GOSUB 1540
680 IF win=1 THEN PEN 15 : LOCATE 15,1 :
PRINT "You win":END
690 mo=1:GOSUB 1210
700 GOSUB 1540
710 IF win =1 THEN PEN 15:LOCATE 15,1:PR
INT "I win":END
720 NEXT goes
730 PEN 15:PRINT "Draw":END
740 :
750 REM
760 REM *****
770 REM * Determine Player's Move *
780 REM *****
790 REM
800 x=1 : y=1
810 FOR i = 1 TO 9
820 INK i,26
830 NEXT i
840 IF x=1 AND y=1 THEN INK 1,1
850 IF x=1 AND y=2 THEN INK 2,1
860 IF x=1 AND y=3 THEN INK 3,1
870 IF x=2 AND y=1 THEN INK 4,1
880 IF x=2 AND y=2 THEN INK 5,1
890 IF x=2 AND y=3 THEN INK 6,1
900 IF x=3 AND y=1 THEN INK 7,1
910 IF x=3 AND y=2 THEN INK 8,1
920 IF x=3 AND y=3 THEN INK 9,1
930 a$=INKEY$ : IF a$="" THEN 930
940 IF a$ <> CHR$(241) THEN GOTO 970

```

```

950   IF y=1 THEN GOTO 930
960   y=y-1 : GOTO 810
970  IF a$ <> CHR$(240) THEN GOTO 1000
980   IF y=3 THEN GOTO 930
990   y=y+1 : GOTO 810
1000  IF a$ <> CHR$(242) THEN GOTO 1030
1010   IF x=1 THEN GOTO 930
1020   x=x-1 : GOTO 810
1030  IF a$ <> CHR$(243) THEN GOTO 1060
1040   IF x=3 THEN GOTO 930
1050   x=x+1 : GOTO 810
1060  IF a$ <> CHR$(224) THEN GOTO 930
1070   IF array(x,y)<>0 THEN SOUND 1,100
      ,100,15:GOTO 930
1080  RETURN
1090  REM *****
1100  REM * Player Win Scan *
1110  REM *****
1120  REM
1130  win =0
1140  FOR i = 1 TO 3
1150   IF array(i,1)=1 AND array(i,2)=1
      AND array(i,3)=1 THEN win = 1
1160   IF array(1,i)=1 AND array(2,i)=1
      AND array(3,i)=1 THEN win = 1
1170  NEXT i
1180  IF array(1,1)=1 AND array(2,2)=1
      AND array(3,3)=1 THEN win = 1
1190  IF array(3,1)=1 AND array(2,2)=1
      AND array(1,3)=1 THEN win = 1
1200  RETURN
1210  win =0
1220  FOR i = 1 TO 3
1230   IF array(i,1)=2 AND array(i,2)=2
      AND array(i,3)=0 THEN array(i,3)
      =2:win=1:RETURN
1240   IF array(i,1)=2 AND array(i,3)=2
      AND array(i,2)=0 THEN array(i,2)
      =2:win=1:RETURN

```




```
1250 IF array(i,2)=2 AND array(i,3)=2
      AND array(i,1)=0 THEN array(i,1)
      =2:win=1:RETURN
1260 IF array(1,i)=2 AND array(2,i)=2
      AND array(3,i)=0 THEN array(3,i)
      =2:win=1:RETURN
1270 IF array(1,i)=2 AND array(3,i)=2
      AND array(1,i)=0 THEN array(1,i)
      =2:win=1:RETURN
1280 IF array(2,i)=2 AND array(3,i)=2
      AND array(1,i)=0 THEN array(1,i)
      =2:win=1:RETURN
1290 NEXT i
1300 IF array(1,1)=2 AND array(2,2)=2
      AND array(3,3)=0 THEN array(3,3)=2
      :win=1:RETURN
1310 IF array(1,1)=2 AND array(2,2)=0
      AND array(3,3)=2 THEN array(2,2)=2
      :win=1:RETURN
1320 IF array(1,1)=0 AND array(2,2)=2
      AND array(3,3)=2 THEN array(1,1)=2
      :win=1:RETURN
1330 IF array(3,1)=2 AND array(2,2)=0
      AND array(1,3)=2 THEN array(2,2)=2
      :win=1:RETURN
1340 IF array(3,1)=2 AND array(2,2)=2
      AND array(1,3)=0 THEN array(1,3)=2
      :win=1:RETURN
1350 IF array(3,1)=0 AND array(2,2)=2
      AND array(1,3)=2 THEN array(3,1)=2
      :win=1:RETURN
1360 FOR i = 1 TO 3
1370 IF array(i,1)=1 AND array(i,2)=1
      AND array(i,3)=0 THEN array(i,3)
      =2:RETURN
1380 IF array(i,1)=1 AND array(i,3)=1
      AND array(i,2)=0 THEN array(i,2)
      =2:RETURN
```

```
1390 IF array(i,2)=1 AND array(i,3)=1
    AND array(i,1)=0 THEN array(i,1)
    =2:RETURN
1400 IF array(1,i)=1 AND array(2,i)=1
    AND array(3,i)=0 THEN array(3,i)
    =2:RETURN
1410 IF array(1,i)=1 AND array(3,i)=1
    AND array(1,i)=0 THEN array(1,i)
    =2:RETURN
1420 IF array(2,i)=1 AND array(3,i)=1
    AND array(1,i)=0 THEN array(1,i)
    =2:RETURN
1430 NEXT i
1440 IF array(1,1)=1 AND array(2,2)=1
    AND array(3,3)=0 THEN array(3,3)=2
    :RETURN
1450 IF array(1,1)=1 AND array(2,2)=0
    AND array(3,3)=1 THEN array(2,2)=2
    :RETURN
1460 IF array(1,1)=0 AND array(2,2)=1
    AND array(3,3)=1 THEN array(1,1)=2
    :RETURN
1470 IF array(3,1)=1 AND array(2,2)=0
    AND array(1,3)=1 THEN array(2,2)=2
    :RETURN
1480 IF array(3,1)=1 AND array(2,2)=1
    AND array(1,3)=0 THEN array(1,3)=2
    :RETURN
1490 IF array(3,1)=0 AND array(2,2)=1
    AND array(1,3)=1 THEN array(3,1)=2
    :RETURN
1500 a1=INT(3*RND(1))+1
1510 b1=INT(3*RND(1))+1
1520 IF array(a1,b1)<>0 THEN GOTO 1500
1530 array(a1,b1)=2 : RETURN
1540 REM draw bits
1550 FOR i = 1 TO 3
1560 FOR j = 1 TO 3
```

```
1570     IF array(i,j)=0 THEN GOTO 1620
1580     FOR k = 1 TO 115 STEP 2
1590         PLOT 146*i,130*j+j-k, 11 +
            array(i,j)
1600     DRAWR -125,0
1610     NEXT k
1620 NEXT j
1630 NEXT i
1640 RETURN
```

Anagram



Rearrange each given jumble of letters into a word in 25 seconds.

```
100 REM *****
110 REM * Anagram *
120 REM *****
130 REM
140 MODE 1
150 BORDER 11
160 INK 0,11
170 INK 1,0
180 INK 2,6,24
190 INK 3,24
200 SPEED INK 10,10
210 REM *****
220 REM * Instructions *
230 REM *****
```

```
240 PEN 1
250 LOCATE 16,1
260 PEN 1
270 PRINT "_____"
280 LOCATE 16,1
290 PEN 3
300 PRINT CHR$(22);"1";"Anagram"
310 PRINT CHR$(22);"0"
320 PEN 1
330 PRINT
340 PRINT "In this game you have to guess the word"
350 PRINT "i have jumbled up."
360 PRINT
370 PRINT "A time limit will be ticking away at"
380 PRINT "the bottom which lasts for 25 seconds."
390 PRINT
400 PRINT "The word will be displayed at"
410 PRINT "the top of the screen and the"
420 PRINT "letters you have guessed below."
430 PRINT "If you finish the word before the"
440 PRINT "the time runs out you will be given a"
450 PRINT "percentage score."
460 PEN 2
470 PRINT
480 PRINT
490 PRINT "      Press any key to start"
500 IF INKEY$="" THEN 500
510 REM *****
520 REM * Choose word *
530 REM *****
```

```

540 RESTORE
550 CLEAR
560 FOR i=1 TO INT(17*RND(0))+1)
570 READ a$:NEXT i
580 word$=a$
590 FOR i=1 TO LEN(word$):part$=part$+"-
"
600 NEXT i
610 ip$=part$
620 FOR i=1 TO LEN(word$)
630 a=INT(LEN(word$)*RND(TIME)+1)
640 IF MID$(ip$,a,1)<>"-" THEN 630
650 ana$=ana$+LOWER$(MID$(word$,a,1))
660 ip$=LEFT$(ip$,a-1)+"*"+RIGHT$(ip$, (L
EN(ip$)-a))
670 NEXT i
680 REM *****
690 REM * set up game screen *
700 REM *****
710 MODE 1
720 SPEED INK 5,5
730 p=1
740 PEN 1
750 INK 0,0
760 INK 1,24
770 INK 2,17
780 INK 3,18
790 LOCATE 15,1
800 PRINT "The word is"
810 LOCATE 20-LEN(ana$)/2,5
820 PRINT ana$
830 FOR i=2 TO 636 STEP 2
840 PLOT i,30,3
850 DRAW i,70
860 NEXT i
870 PLOT 0,28,2:DRAW 0,72:DRAW 639,72:DR
AW 639,28:DRAW 0,28
880 LOCATE 5,9:PRINT "guess :-"

```

```
890 t=LEN(word$)+1
900 d=49
910 c=32
920 PEN 2
930 FOR i=636 TO 2 STEP -0.5
940 IF i<150 THEN INK 3,6,0
950 IF i>150 AND i<300 THEN INK 3,15
960 PLOT i,30,0:DRAW i,70
970 c=c-1:IF c=0 THEN c=d:d=d-1:SOUND 1,
100,2,15
980 a$=INKEY$
990 IF a$="" THEN 1030
1000 GOSUB 1270
1010 LOCATE 14,9
1020 PRINT ip$
1030 NEXT i
1040 REM *****
1050 REM * Game over *
1060 REM *****
1070 SPEED INK 10,10
1080 LOCATE 14,9
1090 PRINT ip$
1100 LOCATE 16,11
1110 PRINT "score =";INT((100/640)*i)
1120 LOCATE 15,1
1130 PRINT "The word was"
1140 LOCATE 20-LEN(ana$)/2,5
1150 PRINT word$
1160 LOCATE 15,14:PEN 3
1170 INK 3,20
1180 FOR i=1 TO 100
1190 SOUND 1,i,1,15
1200 SOUND 2,i,2,15
1210 NEXT i
1220 PRINT "Play again ?"
1230 a$=INKEY$
1240 IF a$="y" THEN 540
1250 IF a$<>"n" THEN GOTO 1230
```

```
1260 END
1270 ip$=LEFT$(ip$,p-1)+a$+RIGHT$(ip$,LE
N(ip$)-p)
1280 p=p+1
1290 IF p=t AND ip$<>LOWER$(word$) THEN
i=1:GOTO 1080
1300 IF p=t THEN GOTO 1080
1310 RETURN
1320 REM *****
****
1330 REM * Word data    this can be chang
ed *
1340 REM *****
****
1350 DATA Amstrad,Rotavator
1360 DATA Establishment
1370 DATA Arnold,Colour
1380 DATA Quantum,Ommadawn
1390 DATA Platinum,Ferrari
1400 DATA Internal,Tommorow
1410 DATA Software,Computer
1420 DATA Addison,Wesley
1430 DATA Sink,Tulip
```



```
180 SYMBOL 92,127,127,127,127,127,127,12
7,127
185 SYMBOL 93,136,17,34,68,136,17,34,68
190 MODE 0
200 PAPER 0:BORDER 0
210 INK 0,0
220 INK 1,21
230 INK 2,25
240 INK 3,3
250 PEN 1
260 REM *****
270 REM * instructions *
280 REM *****
290 LOCATE 7,1
300 PRINT "Hangman"
310 PRINT
320 PEN 3:INK 3,17
330 PRINT "This is the game of"
340 PRINT "Hangman, when asked"
350 PRINT "type in your guess"
360 PRINT "if you have choosen"
370 PRINT "a letter you have"
380 PRINT "already used you "
390 PRINT "will be told"
400 PRINT
410 PEN 2:INK 2,2
420 PRINT "Do you want to input";
430 PRINT "your own word ";
440 INPUT a$
450 IF a$="n" THEN GOTO 490
460 PEN 2:PRINT "Input your word "
470 INPUT word$
480 GOTO 520
490 RESTORE
500 a=INT(14*RND(TIME)+1)
510 FOR i=1 TO a:READ word$:NEXT i
520 MODE 1
```

```

530 PEN 1
540 FOR i=1 TO LEN(word$)
550 p$="-":IF MID$(word$,i,1)=" " THEN p
    $=" "
560 find$=find$+p$
570 NEXT i
580 LOCATE 2,4:PRINT "Letters Used"
581 PEN 2
590 LOCATE 2,5:PRINT "]]]]]]]]]]]]]]]"
600 LOCATE 2,6:PRINT "]"                "]"
610 LOCATE 2,7:PRINT "]]]]]]]]]]]]]]]"
615 PEN 1
620 LOCATE 2,2:PRINT find$
630 LOCATE 2,10:PRINT SPACE$(16)
640 f=0
650 REM *****
660 REM * Input a letter *
670 REM *****
680 LOCATE 2,10:INPUT "letter ?";l$
690 IF LEN(l$)=1 THEN GOTO 740
700 LOCATE 2,10:PRINT SPACE$(28):GOTO 68
    0
710 REM *****
720 REM * Check for a dud *
730 REM *****
740 IF l$<"a" OR l$>"z" THEN GOTO 620
750 FOR i=1 TO LEN (used$)
760 IF l$=MID$(used$,i,1) THEN GOSUB 194
    0:GOTO 620
770 NEXT i
780 FOR i=1 TO LEN (find$)
790 IF l$=MID$(find$,i,1) THEN GOSUB 19
    4:GOTO 620
800 NEXT i
810 FOR i=1 TO LEN (word$)
820 IF LOWER$(MID$(word$,i,1))<> l$ THEN
    850
830 find$=LEFT$(find$,i-1)+l$+RIGHT$(fin
    d$,LEN(word$)-i)

```

```

840 f=1
850 NEXT i
860 IF find$=LOWER$(word$) THEN 2040
870 IF f=1 THEN GOTO 620
880 used$=used$+1$
890 p=p+1
900 REM *****
910 REM * Draw part of man *
920 REM *****
930 ON p GOSUB 1010,1060,1170,1240,1320,
1370,1420,1620,1710,1780
940 PEN 1
950 IF LEN (used$)=10 THEN 2210
960 LOCATE 3,6:PRINT used$
970 GOTO 620
980 REM *****
990 REM * Draw base *
1000 REM *****
1010 PEN 3:FOR i=1 TO 11:LOCATE 20+i,23:
PRINT CHR$(143);:NEXT i
1020 RETURN
1030 REM *****
1040 REM * Draw base supports *
1050 REM *****
1060 PEN 3:FOR i=1 TO 4:LOCATE 21+i,23-i
:PRINT CHR$(214):NEXT i
1070 FOR i=1 TO 3:LOCATE 22+i,23-i:PRINT
CHR$(212):NEXT i
1080 LOCATE 26,19:PRINT CHR$(143)
1090 FOR i=1 TO 4
1100 LOCATE 26+i,18+i:PRINT CHR$(215):NE
XT i
1110 FOR i=1 TO 3
1120 LOCATE 26+i,19+i:PRINT CHR$(213):NE
XT i
1130 RETURN
1140 REM *****
1150 REM * Draw upright *
1160 REM *****

```

```

1170 PEN 3:FOR i=22 TO 7 STEP -1
1180 LOCATE 26,i:PRINT CHR$(143)
1190 NEXT i
1200 RETURN
1210 REM *****
1220 REM * Draw beam support *
1230 REM *****
1240 PEN 3:FOR i=1 TO 3
1250 LOCATE 26+i,10-i:PRINT CHR$(212):NE
XT i
1260 FOR i=1 TO 2
1270 LOCATE 26+i,9-i:PRINT CHR$(214):NEX
T i
1280 RETURN
1290 REM *****
1300 REM * Draw beam *
1310 REM *****
1320 PEN 3:FOR i=26 TO 35:LOCATE i,6:PRI
NT CHR$(143);:NEXT i
1330 RETURN
1340 REM *****
1350 REM * Draw rope *
1360 REM *****
1370 PEN 2:LOCATE 35,7:PRINT CHR$(149):L
OCATE 35,8:PRINT CHR$(149)
1380 RETURN
1390 REM *****
1400 REM * Draw head *
1410 REM *****
1420 FOR i=1 TO 3:LOCATE 33+i,9
1430 PRINT CHR$(218);
1440 LOCATE 33+i,10:PRINT CHR$(208):NEXT
i
1450 LOCATE 33,10:PRINT CHR$(204)
1460 LOCATE 37,10:PRINT CHR$(205)
1470 LOCATE 33,11:PRINT CHR$(211)
1480 LOCATE 33,12:PRINT CHR$(211)
1490 LOCATE 37,11:PRINT CHR$(209)

```

```
1500 LOCATE 37,12:PRINT CHR$(209)
1510 LOCATE 33,13:PRINT CHR$(205):LOCATE
 37,13:PRINT CHR$(204)
1520 FOR i=1 TO 3:LOCATE 33+i,13:PRINT C
HR$(210):NEXT i
1530 LOCATE 34,11:PRINT CHR$(230):LOCATE
 36,11:PRINT CHR$(230)
1540 LOCATE 35,12:PRINT CHR$(178)
1550 PRINT CHR$(22);"1"
1560 LOCATE 34,14:PRINT CHR$(211):LOCATE
 36,14
1570 PRINT CHR$(209):PRINT CHR$(22);"0"
1580 RETURN
1590 REM *****
1600 REM * Draw body *
1610 REM *****
1620 PEN 1:FOR i=0 TO 4:LOCATE 33+i,14:P
RINT CHR$(143);:NEXT i
1630 FOR i=0 TO 6:LOCATE 32+i,15:PRINT C
HR$(143);:NEXT i
1640 FOR i=0 TO 6:LOCATE 32+i,16:PRINT C
HR$(143);:NEXT i
1650 FOR i=0 TO 6:LOCATE 32+i,17:PRINT C
HR$(143);:NEXT i
1660 LOCATE 32,14:PRINT CHR$(214):LOCATE
 38,14:PRINT CHR$(215)
1670 RETURN
1680 REM *****
1690 REM * Draw arms *
1700 REM *****
1710 PEN 2:FOR i=15 TO 17:LOCATE 32,i:PR
INT CHR$(91)
1720 LOCATE 38,i:PRINT CHR$(92)
1730 NEXT i
1740 RETURN
1750 REM *****
1760 REM * Draw legs *
1770 REM *****
```

```

1780 PEN 1:FOR j=18 TO 22
1790 FOR i=1 TO 5
1800 IF j>19 AND i=3 THEN 1820
1810 LOCATE 32+i,j:PRINT CHR$(206)
1820 NEXT i:NEXT j
1830 RETURN
1840 REM *****
***
1850 REM * Word data Words can be change
d *
1860 REM *****
***
1870 DATA Amstrad,Discovery
1880 DATA Newtech,Addison Wesley
1890 DATA Sierra,Escort
1900 DATA Monitor,Printer
1910 DATA Television,Telephone
1920 DATA Mark,Vector Twin Turbo
1930 DATA Submerge,Daffodil
1940 PEN 2:PAPER 3
1950 SOUND 1,200,50,12,2,2
1960 LOCATE 2,10:PRINT "Letter already u
sed"
1970 FOR i=1 TO 1000:NEXT i
1980 PEN 1:PAPER 0
1990 LOCATE 2,10:PRINT SPACE$(19)
2000 RETURN
2010 REM *****
2020 REM * Saved          phewwww! *
2030 REM *****
2040 CLS
2050 FOR i=1 TO 25
2060 c=c+1:IF c=4 THEN c=1
2070 PEN c
2080 INK c,RND(TIME)*29
2090 LOCATE 15,i:PRINT "You saved me"
2100 SOUND 1,i*10-(i*3.4),7,10,1,1
2105 NEXT i

```

```

2110 FOR i=25 TO 2 STEP -1
2120 c=c+1:IF c=4 THEN c=1
2130 INK c,RND(TIME)*29
2140 LOCATE 15,i:PRINT SPACE$(12)
2150 SOUND 1,i*10-(i*3.3),7,10,1,1
2160 NEXT i
2170 GOTO 2560
2180 REM *****
2190 REM * Died          aaahhh *
2200 REM *****
2210 LOCATE 35,12:PRINT CHR$(230)
2220 LOCATE 34,11:PRINT CHR$(231)
2230 LOCATE 36,11:PRINT CHR$(231)
2240 FOR i=0 TO 200 STEP 2.3
2250 SOUND 1,i,3,10,1,1
2260 NEXT i
2270 REM *****
2280 REM * R.I.P. *
2290 REM *****
2300 CLS
2310 PEN 2:INK 2,21
2320 LOCATE 1,14:PRINT STRING$(40,143)
2330 PEN 3:INK 3,13
2340 FOR i=13 TO 4 STEP -1
2350 LOCATE 14,i:PRINT STRING$(10,143)
2360 LOCATE 14,I:PRINT CHR$(91):NEXT i
2370 LOCATE 15,3:PRINT STRING$(8,143)
2380 LOCATE 16,2:PRINT STRING$(6,143)
2390 PAPER 3:PEN 1:LOCATE 17,6:PRINT "R"
2400 LOCATE 19,6:PRINT "I"
2410 LOCATE 21,6:PRINT "P"
2420 PEN 0:PAPER 3
2430 LOCATE 15,3:PRINT CHR$(204)
2440 LOCATE 16,2:PRINT CHR$(204)
2450 PAPER 0
2460 PEN 3
2470 LOCATE 14,3:PRINT CHR$(214)
2480 LOCATE 15,2:PRINT CHR$(214)

```




```
2490 LOCATE 23,3:PRINT CHR$(215)
2500 LOCATE 22,2:PRINT CHR$(215)
2510 FOR i=0 TO 640 STEP 4:PLOT i,190,2
2520 DRAW i+INT(20*RND(TIME)-10),200+15+
RND(TIME)*15
2530 NEXT i
2540 LOCATE 14,16:PRINT "The word was"
2550 LOCATE 20-(LEN(word$)/2),18:PRINT w
ord$
2560 PEN 1:INK 1,6
2570 REM *****
2580 REM * Game over *
2590 REM *****
2600 LOCATE 15,20:PRINT "Play again?"
2610 a$=INKEY$:IF a$="" THEN GOTO 2610
2620 IF a$="y" THEN RUN
2630 IF a$<>"n" THEN GOTO 2610
2640 END
```

Chapter 8

TWO-USER GAMES

Moon Base



A two-player game to test your reactions (and the Amstrad's keyboard).
Player 1 must try to destroy the advancing alien spacecraft (Player 2)
before they overrun the moon base.

```
100 REM *****
110 REM * Moonbase *
120 REM *****
130 ENT 1,10,-1,1
140 ENV 2,5,3,1
150 ENT 2,100,3,1
160 b1$=CHR$(143)
170 p1=9:p2=0
180 c1$=SPACE$(40)
190 MODE 1:PAPER 0:INK 0,0
200 INK 1,12
210 LOCATE 16,1:PRINT"Moonbase"
```

```
220 PRINT:PEN 2
230 PRINT "Player 1"
240 PEN 1
250 PRINT "You have been on a mission to
  saturn"
260 PRINT "to collect soil samples for r
  esearch."
270 PRINT "While you were away on your m
  ission"
280 PRINT "civil war broke out and your
  landing"
290 PRINT "pad has been captured."
300 PRINT "you have a limited amount of
  fuel"
310 PRINT "with which to land."
320 PRINT
330 PRINT "Your controls are as follows"
340 PRINT
350 PRINT CHR$(154);CHR$(246);" Right"
360 PRINT
370 PRINT "!  down"
380 PRINT CHR$(245)
390 PEN 3:PRINT
400 PRINT "Press a key for player 2's in
  structions"
410 IF INKEY$="" THEN GOTO 410
420 PEN 1
430 PRINT "          Moonbase"
440 PRINT:PEN 2
450 PRINT "Player 2"
460 PEN 1
470 PRINT "as one of the gunners invadin
  g"
480 PRINT "the Moonbase your objective i
  s"
490 PRINT "to shoot down the lander"
500 PRINT
510 PRINT "Your controls are as follows"
```

```

520 PRINT
530 PRINT "A ";CHR$(247);CHR$(154);"#";
540 PRINT CHR$(154);CHR$(246);" D"
550 PRINT "F fires the cannon"
560 PRINT:PEN 3
570 PRINT "press any key to start game"
580 FOR i=1 TO 12:PRINT:NEXT i
590 IF INKEY$="" THEN GOTO 590
600 MODE 1:INK 0,0:BORDER 0
610 PAPER 0
620 INK 1,2
630 INK 2,24
640 GOSUB 1140
650 dx=10
660 f=0
670 GOSUB 1480
680 PAPER 3:LOCATE 1,1:PRINT c1$
690 PAPER 0
700 x=3:y=4
710 REM *****
720 REM * Start main loop *
730 REM *****
740 a$=INKEY$
750 IF a$="" THEN GOTO 810
760 IF a$=CHR$(243) THEN x=x+1
770 IF a$=CHR$(241) THEN y=y+1
780 IF a$="a" AND dx>2 THEN dx=dx-1
790 IF a$="d" AND dx<29 THEN dx=dx+1
800 IF a$="f" AND f=0 THEN f=1:fx=dx+2:f
y=23
810 IF f=0 THEN GOTO 900
820 LOCATE fx,fy:PRINT " "
830 fx=fx+1:fy=fy-1:IF fy=3 THEN f=0:GOT
O 900
840 ff=TEST((fx*16),(400-fy*16)+12)
850 SOUND 1,fy*2,10,10,1,1
860 IF ff=2 THEN f=0:GOTO 900
870 IF ff=1 THEN GOTO 1420

```

```

880 PEN 3
890 LOCATE fx,fy:PRINT CHR$(194)
900 IF a$="" THEN 740
910 IF x=35 AND y=21 THEN 1370
914 REM *****
915 REM * Hit anything *
916 REM *****
920 IF TEST((x*16+8),(400-y*16)-32)=2 TH
EN GOTO 1320
930 IF TEST(x*16+20,400-y*16-32)=2 THEN
GOTO 1320
940 IF a$=CHR$(243) AND TEST((x*16+25),(
400-y*16)-24)=2 THEN GOTO 1320
950 GOSUB 990
960 GOSUB 1070
970 GOTO 740
980 REM *****
990 REM * Draw space ship *
1000 REM *****
1010 PEN 1
1020 LOCATE x-1,y:PRINT "      "
1030 LOCATE x-1,y+1:PRINT " ";CHR$(222);
CHR$(207);CHR$(223)
1040 LOCATE x-1,y+2:PRINT " ";CHR$(155);
" ";CHR$(155)
1050 RETURN
1060 REM *****
1070 REM * Draw gunner *
1080 REM *****
1090 PEN 3
1100 LOCATE dx,23:PRINT " ";CHR$(132);CH
R$(204);" "
1110 LOCATE dx,24:PRINT " ";CHR$(143);CH
R$(215);" "
1120 RETURN
1130 REM *****
1140 REM * Draw moonbase *
1150 REM *****

```

```


1160 PEN 2
1170 LOCATE 34,21:PRINT "    ";CHR$(135)
;CHR$(135)
1180 LOCATE 34,22:PRINT "    ";CHR$(135)
;CHR$(135)
1190 LOCATE 34,23:PRINT "    ";CHR$(135)
;CHR$(135)
1200 LOCATE 33,24:PRINT b1$;b1$;b1$;b1$;
b1$;b1$;
1210 LOCATE 33,24:PRINT CHR$(222)
1220 LOCATE 34,23:PRINT CHR$(222)
1230 FOR i=2 TO 25
1240 LOCATE 40,i:PRINT CHR$(143);
1250 LOCATE 1,i:PRINT CHR$(143);
1260 NEXT i
1270 FOR i=1 TO 40:LOCATE i,25:PRINT b1$
;:NEXT i
1280 RETURN
1290 REM *****
1300 REM * Midgame messages *
1310 REM *****
1320 PAPER 3:PEN 1:LOCATE 1,1:PRINT c1$
1330 LOCATE 11,1:PRINT "player 1 crashed
"
1340 p2=p2+1
1350 IF p2=10 THEN GOTO 1680
1360 GOTO 1550
1370 PAPER 3:PEN 1:LOCATE 1,1:PRINT c1$
1380 LOCATE 11,1:PRINT "player 1 landed"
1390 p1=p1+1
1400 IF p1=10 THEN GOTO 1680
1410 GOTO 1550
1420 PAPER 3:PEN 1:LOCATE 1,1:PRINT c1$
1430 LOCATE 11,1:PRINT "player 2 hit pla
yer 1"
1440 p2=p2+1
1450 IF p2=10 THEN GOTO 1680
1460 SOUND 1,i,100,10,2,2,9

```



```
1470 GOTO 1550
1480 PAPER 2
1490 PEN 3
1500 LOCATE 1,2
1510 PRINT c1$
1520 LOCATE 2,2:PRINT"player 1 :-";p1
1530 LOCATE 20,2:PRINT "player 2 :-";p2
1540 RETURN
1550 GOSUB 1480
1560 FOR i=1 TO 1000:NEXT i
1570 PAPER 3
1580 LOCATE 1,1:PRINT c1$
1590 PAPER 0
1600 f=0
1610 LOCATE 1,1:FOR i=1 TO 26:PRINT CHR$(11);:NEXT i
1620 GOSUB 1140
1630 GOSUB 1480
1640 GOTO 690
1650 REM *****
1660 REM * Game over *
1670 REM *****
1680 MODE 1:PAPER 0:INK 0,0
1690 INK 1,14
1700 LOCATE 15,10:PRINT "Game over"
1710 IF p1=10 THEN LOCATE 14,12:PRINT "P
layer 1 won"
1720 IF p2=10 THEN LOCATE 14,12:PRINT "P
layer 2 Won"
1730 LOCATE 13,16:PRINT "Play again ?"
1740 a$=INKEY$
1750 IF a$="y" THEN RUN
1760 IF a$<>"n" THEN GOTO 1740
```

Light-bikes



Entrap your opponent in your plasma beams before he gets you.

```
100 REM *****
110 REM * Light bikes *
120 REM *****
130 MODE 1:INK 1,6:PEN 1
135 BORDER 0
140 PAPER 0:INK 0,2
150 INK 2,24
160 INK 3,0
180 SYMBOL AFTER 91
190 SYMBOL 91,7,24,32,192,128,135,72,48
200 SYMBOL 92,224,24,4,3,1,225,18,12
210 SYMBOL 93,0,7,31,63,255,252,48,0
220 SYMBOL 94,0,240,248,252,255,63,12,0
```

```

230 REM *****
240 REM * Instructions *
250 REM *****
260 m$="Light cycles":p=15:q=1:GOSUB 121
0
270 m$="You are about to be placed into
the"
280 p=1:q=3:GOSUB 1210
290 m$="LIGHT CYCLE game grid.."
300 p=1:q=4:GOSUB 1210
310 m$="Try to surround the other player
"
320 p=1:q=5:GOSUB 1210
330 m$="with the trail left by your ligh
t"
340 p=1:q=6:GOSUB 1210
350 m$="cycle."
360 p=1:q=7:GOSUB 1210
370 m$="Player 1 uses keys <2> <4> <6> <
8>"
380 p=1:q=9:GOSUB 1210
390 m$="Respectivly on the numeric key p
ad"
400 p=1:q=10:GOSUB 1210
410 m$="Player 2 uses a joystick plugged
into"
420 p=1:q=12:GOSUB 1210
430 m$="port 1"
440 p=1:q=13:GOSUB 1210
450 m$="Press any key to enter GAME GRID
"
460 p=1:q=17:GOSUB 1210
470 IF INKEY$="" THEN 470
480 REM *****
490 REM * Draw game grid *
500 REM *****
510 MODE 0
520 x1=0:y1=180:yd1=0:xd1=4:x=0:y=200:yd
=0:xd=4

```

```

530 c=8
540 FOR i=8 TO 14:INK i,13:NEXT i
550 INK 15,6
560 INK 7,6:REM * grid lines
570 INK 1,2
580 INK 2,24
590 FOR i=0 TO 400 STEP 2
600 PLOT 0,i,c:DRAW 640,i
610 c=c+1:IF c=16 THEN c=8
620 NEXT i
630 FOR i=0 TO 640 STEP 20:PLOT i,0:DRAW
    i,400,7
640 NEXT i
650 a=8:b=9
660 GOTO 800
670 a=a+1:IF a=16 THEN a=8
680 b=b+1:IF b=16 THEN b=8
690 INK a,13:INK b,6
700 RETURN
710 REM *****
720 REM * Main loop *
730 REM *****
740 a$=INKEY$
750 GOSUB 670
760 IF a$="8" THEN yd=2:xd=0
770 IF a$="2" THEN yd=-2:xd=0
780 IF a$="6" THEN xd=4:yd=0
790 IF a$="4" THEN xd=-4:yd=0
800 y=y+yd
810 x=x+xd
820 IF TEST(x,y)<7 THEN 930
830 js=JOY(0)
840 IF js=1 THEN yd1=2:xd1=0
850 IF js=2 THEN yd1=-2:xd1=0
860 IF js=4 THEN yd1=0:xd1=-4
870 IF js=8 THEN yd1=0:xd1=4
880 x1=x1+xd1:y1=y1+yd1
890 IF TEST(x1,y1)<7 THEN 1070
900 PLOT x1,y1,2

```


```
910 PLOT x,y,1
920 GOTO 740
930 FOR i=25 TO 0 STEP -0.5
940 FOR j=1 TO 10:NEXT j
950 INK 1,i:SOUND 1,i*2.5,1,15
960 NEXT i
970 REM *****
980 REM * Light cycle 1 derezzed *
990 REM *****
1000 MODE 1
1010 PEN 1:INK 1,2
1020 m$="Light cycle 1 derezzed":p=10:q=
10
1030 INK 1,6:INK 2,24:INK 3,0
1040 GOSUB 1210
1050 p2=p2+1:IF p2=10 THEN 1390
1060 GOTO 1530
1070 FOR i=25 TO 0 STEP -0.5
1080 FOR j=1 TO 10:NEXT j
1090 INK 2,i:SOUND 1,i*2.5,1,15
1100 NEXT i
1110 REM *****
1120 REM * Light cycle 2 derezzed *
1130 REM *****
1140 MODE 1
1150 INK 0,2
1160 m$="Light cycle 2 derezzed":p=10:q=
10
1170 INK 1,6:INK 2,24:INK 3,0
1180 GOSUB 1210
1190 p1=p1+1:IF p1=10 THEN 1390
1200 GOTO 1530
1210 FOR i=1 TO LEN(m$)
1220 LOCATE p,q
1230 p=p+1
1240 PEN 1
1250 PRINT MID$(m$,i,1);
1260 SOUND 1,400+p+q,7,15,10,10,1
```

```

1270 PEN 2
1280 PRINT CHR$(93);CHR$(94);CHR$(8);CHR
$(8);CHR$(22);"1";
1290 PEN 3
1300 PRINT CHR$(91);CHR$(92);CHR$(22);"0
"
1310 FOR j=1 TO 10:NEXT j
1320 NEXT i
1330 FOR j=1 TO 400:NEXT j
1340 LOCATE p,q:PRINT " "
1350 RETURN
1360 REM *****
1370 REM * Game over *
1380 REM *****
1390 MODE 1
1400 m$="Game Over"
1410 p=15:q=3
1420 GOSUB 1210
1430 IF p1=10 THEN m$="Player 1 Won"
1440 IF p2=10 THEN m$="Player 2 Won"
1450 p=14:q=10
1460 GOSUB 1210
1470 m$="Play again"
1480 p=15:q=14:GOSUB 1210
1490 a$=INKEY$
1500 IF a$="y" THEN RUN
1510 IF a$="n" THEN p=5:q=3:m$="Light Cy
cles escaping game grid":GOSUB 1210:END
1520 GOTO 1490
1530 MODE 1
1540 m$=" Player one Score "+STR$(p1)
1550 p=3:q=5:GOSUB 1210
1560 m$=" Player two Score "+STR$(p2)
1570 p=3:q=10:GOSUB 1210
1580 FOR i=1 TO 3000:NEXT i:GOTO 510

```

Shoot-out



- At high noon you and your partner are going to have a shoot-out at Amstrad Gulch. You will have a choice of where you're going to have the fight. It will be either
1. behind a row of cacti
 2. behind a broken chuck wagon
 3. behind a broken chuck wagon with lots of cacti scattered about.

```
100 REM *****
110 REM * Shoot-out *
120 REM *****
130 MODE 1: BORDER 12
140 DEF FNr (r)=INT (r*RND (TIME)+1)
150 DEF FNz1 (z)=z*16+(RND (TIME)*2)
160 DEF FNz2 (z)=408-(z*16)
170 INK 0,25: INK 1,9: INK 2,3: INK 3,4
180 cr$=" $ $ $ $ $": al$=" "+CHR$(248)
190 al$=al$+al$+al$+al$+al$
```

```

200 c11$=CHR$(200)+CHR$(201)
210 c12$=CHR$(202)+CHR$(203)
220 c21$=CHR$(204)+CHR$(205)
230 c22$=CHR$(206)+CHR$(207)
240 dd$=CHR$(208)+CHR$(209)+CHR$(210)
250 p2=5:p1=5:SYMBOL AFTER 35
260 SYMBOL 36,24,24,126,126,24,24,24,24
270 SYMBOL 37,255,255,255,255,255,255,25
5,255
280 SYMBOL 91,1,3,3,3,3,3,67,227
290 SYMBOL 92,0,128,128,128,136,156,156,
156
300 SYMBOL 93,227,227,227,255,255,127,3,
3
310 SYMBOL 94,252,252,248,128,128,128,12
8,128
320 SYMBOL 200,1,5,7,1,3,1,7,11
330 SYMBOL 201,64,208,240,192,192,192,24
0,248
340 SYMBOL 202,27,15,7,3,3,6,6,14
350 SYMBOL 203,240,224,224,96,96,96,96,2
24
360 SYMBOL 204,2,11,15,3,3,3,15,31
370 SYMBOL 205,128,160,224,128,192,128,2
24,208
380 SYMBOL 206,15,7,7,6,6,6,6,7
390 SYMBOL 207,216,224,192,192,192,96,96
,112
400 SYMBOL 208,0,0,0,32,115,255,255,123
410 SYMBOL 209,0,0,0,0,240,255,255,255
420 SYMBOL 210,0,0,0,0,0,17,255,255
430 c$=CHR$(91)+CHR$(92)+CHR$(10)+CHR$(8
)+CHR$(8)
440 c$=c$+CHR$(93)+CHR$(94):PEN 1
450 LOCATE 15,1:PRINT "Shoot out":PRINT
460 PRINT "At high noon you and your par
tner are"
470 PRINT "Going to have a shoot out at
AMSTRAD"

```



```

480 PRINT "GULCH,You will have a choice
of where"
490 PRINT "you are going to have the fig
ht."
500 PRINT "They will be either...."
510 PRINT "                Controls"
520 PRINT "Player 1 uses the Cursor cont
rols"
530 PRINT "and the copy key"
540 PRINT "Player 2 uses keys W X A D"
550 PRINT "and 's'"
560 PRINT"                Options"
570 PRINT " <1> cacti in the middle"
580 PRINT " <2> a chuck wagon in the mid
dle"
590 PRINT " <3> random cacti and a chuck
wagon":PRINT
600 PRINT "                Input choice"
610 z$=INKEY$:IF z$<"1"OR z$>"3"THEN 610
620 MODE 1:x2=35:y2=15:x1=5:y1=15
630 LOCATE 1,1:PRINT LEFT$(a1$,p1*2);LEF
T$(cr$, (10-p1*2))
640 LOCATE 20,1:PRINT LEFT$(a1$,p2*2);LE
FT$(cr$, (10-p2*2))
650 ON VAL(z$) GOTO 660,670,750
660 FOR i=4 TO 22 STEP 4:LOCATE 20,i:PEN
1:PRINT c$:NEXT i:GOTO 790
670 PEN 2:LOCATE 19,11:PRINT CHR$(194);
680 PRINT CHR$(154);CHR$(195)
690 LOCATE 19,12:PRINT CHR$(149);
700 PRINT " ";CHR$(149)
710 LOCATE 18,13:PRINT CHR$(149);STRING$(
3,143);CHR$(149)
720 LOCATE 18,14:PRINT CHR$(151);STRING$(
3,143);CHR$(157)
730 LOCATE 18,15:PRINT CHR$(149);" ";
740 PRINT CHR$(149):GOTO 790
750 PRINT CHR$(22);"1"

```

```

760 FOR i=1 TO 10:a=FNr(31)+3:b=FNr(18)+
2
770 LOCATE a,b:PEN 1:PRINT c$:NEXT i
780 PRINT CHR$(22);"0":GOTO 670
790 PEN 3:LOCATE x2,y2:PRINT c11$
800 LOCATE x2,y2+1:PRINT c12$
810 LOCATE x1,y1:PRINT c21$
820 LOCATE x1,y1+1:PRINT c22$
830 INK 1,9:INK 2,3:INK 3,7:INK 0,25
840 a$=INKEY$:IF a$="" THEN 930
850 IF a$<CHR$(200) THEN 940
860 IF a$=CHR$(224) AND f2=0 THEN f2=1:t
x2=FNz1(x2):ty2=FNz2(y2)
870 LOCATE x2,y2:PRINT " "
880 LOCATE x2,y2+1:PRINT " "
890 x2=x2+(a$=CHR$(242) AND x2>23)-(a$=C
HR$(243) AND x2<38)
900 y2=y2+(a$=CHR$(240) AND y2>4)-(a$=CH
R$(241) AND y2<22)
910 LOCATE x2,y2:PRINT c11$
920 LOCATE x2,y2+1:PRINT c12$
930 IF f2=1 THEN GOSUB 1040
940 IF a$>CHR$(123) OR a$="" THEN 1020
950 IF a$="s" AND f1=0 THEN f1=1:tx1=FNz
1(x1):ty1=FNz2(y1)
960 LOCATE x1,y1:PRINT " "
970 LOCATE x1,y1+1:PRINT " "
980 x1=x1+(a$="a" AND x1>1)-(a$="d" AND
x1<16)
990 y1=y1+(a$="w" AND y1>4)-(a$="x" AND
y1<22)
1000 LOCATE x1,y1:PRINT c21$
1010 LOCATE x1,y1+1:PRINT c22$
1020 IF f1=1 THEN GOSUB 1100
1030 GOTO 840
1040 PLOT tx2,ty2,0:tx2=tx2-8
1050 IF tx2<0 THEN f2=0
1060 tt=TEST(tx2,ty2)

```

```

1070 IF tt=1 OR tt=2 THEN f2=0:sx=tx2:sy
=ty2:GOSUB 1160:RETURN
1080 IF tt=3 THEN dx=x1:dy=y1:GOTO 1200
1090 PLOT tx2,ty2,2:RETURN
1100 PLOT tx1,ty1,0:tx1=tx1+8
1110 IF tx1>640 THEN f1=0
1120 tt=TEST(tx1,ty1)
1130 IF tt=1 OR tt=2 THEN f1=0:sx=tx1:sy
=ty1:GOSUB 1160:RETURN
1140 IF tt=3 THEN dx=x2:dy=y2:GOTO 1200
1150 PLOT tx1,ty1,2:RETURN
1160 FOR i=1 TO 15
1170 PLOT sx+FNr(10)-5,sy+FNr(10)-5,0
1180 SOUND 1,3000+i,3,15,15,15,FNr(15)
1190 NEXT i:PLOT sx,sy,0:RETURN
1200 sx=dx:sy=dy:GOSUB 1160
1210 LOCATE dx,dy:PRINT " "
1220 LOCATE dx,dy+1:PRINT " "
1230 FOR i=400-dy*16 TO 424 STEP 4
1240 TAG:PLOT dx*16,i,3:PRINT dd$;
1250 PLOT dx*16,i-16:PRINT " ";
1260 TAGOFF:NEXT i
1270 IF dx*16>320 THEN p2=p2-1
1280 IF dx*16<320 THEN p1=p1-1
1290 IF p2=0 OR p1=0 THEN 1320
1300 LOCATE 1,1:PRINT LEFT$(a1$,p1*2);LE
FT$(cr$, (10-p1*2))
1310 LOCATE 20,1:PRINT LEFT$(a1$,p2*2);L
EFT$(cr$, (10-p2*2)):GOTO 1310
1320 MODE 1:PEN 1:LOCATE 15,3
1330 PRINT "Game over":PEN 2:LOCATE 14,7
1340 IF p2=0 THEN PRINT "Player 1 Won"
1350 IF p1=0 THEN PRINT "Player 2 Won"
1360 LOCATE 15,11:PRINT "Play again"
1370 a$=INKEY$:IF a$="y" THEN RUN

```


Programs to Delight and Dazzle!

An Amstrad CPC464 Compendium invites you, the keen computer hobbyist, to key in and play a fascinating selection of thirty games on the Amstrad CPC464.

Each game is presented with a complete listing in LOCOMOTIVE BASIC, a helpful introductory description and a screenshot.

There is something to appeal to the traditional and ultramodern computer gamester alike.

Colourful graphics and art programs are included and you are introduced to a drawing program that allows you to 'paint' detailed pictures of your own creation.

The authors set out with the intention of encouraging you to develop programming skills, allowing you to experiment with the programs in the book or even to write your own original programs. But the emphasis remains strictly on the fun you can have exploring the powerful facilities of your own Amstrad CPC464!



A Newtech Production for
Addison-Wesley Publishing Company

GB £ NET +007.95

ISBN 0-201-15439-0



9 780201 154399

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMERICAN STAINLESS STEEL COMPANY

AMSTRAD

CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.